
Preface



Perl is arguably the most popular scripting language in use today. It
is used for a wide variety of tasks, including file processing,
system administration, web programming, and database connectivity.
Early Perl users had to be content with command-line interfaces or
full-screen interfaces using Curses or similar systems, but the
splitting-off of the Tk widget library from the Tcl language opened a
whole new world to Perl. Perl programmers could now easily create
graphical interfaces for their programs using Tk's flexible and
friendly widget set and, with little effort, those programs could be
made to work across Windows and Unix platforms.

The relatively recent advent of the web browser would seem to have
made the Tk interface obsolete. CGI programs are almost inherently
cross-platform and provide many of the same widgets as Tk (this
includes Menus, Buttons, text entry fields, and so on). However, the
inherent statelessness of the Web makes it difficult to write some
programs for it. Perl/Tk provides a richer widget set than that
available to the CGI programmer. Server push and client pull try to
get around some of these limitations, while JavaScript fills in other
gaps, but the fact is, the user experience still falls short in many
instances. It is for precisely this reason that Perl/Tk continues to
flourish.

The Tk module gives the Perl programmer full access to the powerful
Tk widget set. This rich and diverse library, like Perl itself, makes
the easy things easy and the hard things possible. Easy things
include designing graphical interfaces with Buttons, Checkbuttons,
Menus, and text entry fields—all of which you will learn about
in the first half of this book. The second half of the book contains
more advanced topics, such as creating custom widgets, interprocess
communication, images, animation, and key bindings. The goal of this
book is to take you from Tk neophyte to Tk expert.

History of This Book



This book evolved from the book Learning
Perl/Tk by Nancy Walsh (O'Reilly). While the
response to that book made it clear that there were many avid and
loyal Perl/Tk users, readers also let us know that they wished the
book showed advanced techniques, covered some nonstandard widgets,
and included more extended examples of Perl/Tk programming.

So Nancy teamed up with Steve Lidie, already a seasoned Perl/Tk
programmer and author, and together they doubled the size of the book
and changed its title to befit its expansion. The baby emu on the
cover of Learning Perl/Tk grew into the adult
emu on the cover of this book. Programmers who are new to Perl/Tk can
still learn from this book, but as the title says, this book will
also turn you into a master.



What You Should Already Know



To get the most out of this book, you should already know the basics
of programming in Perl, Version 5. You don't have to be a Perl
guru to learn Perl/Tk, but it will help if you feel comfortable with
the language.[1]
         
Perl/Tk utilizes the object-oriented features available in Perl 5, so
even if you don't completely understand OO programming, you
should be able to recognize it when you see it. The only other thing
you'll need is prior knowledge of other graphical user
interfaces (GUIs) and your opinions on them. This helps when deciding
what features to include in your own applications. Take a look at the
word processor you use on your PC, your web browser, or any program
that has buttons and scrollbars and accepts both mouse and keyboard
input.

In this book, we'll be covering each basic widget and all its
associated options in detail. You'll learn how to make a window
look the way you want it to look. You'll also learn how to make
a window user-friendly and attractive. Other important topics include
image creation and manipulation, interprocess communications, and
mega-widget details. We also take a look inside the Tk event loop,
encompassing events, bindings, and callbacks.

To complement the examples and code snippets, you will find complete
programs scattered throughout the book and in Appendix C. These range from RPN calculators and LWP web
clients to Robot Control programs written in Perl and Tcl.

If you want to know more about Perl in general, you should read
Learning Perl, Programming Perl,
Advanced Perl Programming, and The Perl
Cookbook, which are all also published by O'Reilly
& Associates, Inc. From other publishers,
you might also try Object Oriented Perl
(Manning), Elements of Programming in Perl
(Manning), and Effective Perl Programming
(Addison Wesley). There are also numerous FAQs and documents
available on the Web.



[1] Here's the laundry list of things
you should at least recognize: hashes, arrays, subroutines, and their
anonymous versions, as well as
$
                  _ and
@
                  _.



What's in This Book



Here is the breakdown of what we cover in this book:
	
                  Chapter 1
               
	The first chapter contains some interesting history about the Tk
module, introductory comments, and the obligatory Hello World
program.


	
                  Chapter 2
               
	Geometry management is probably the most important concept in using
Perl/Tk. It determines how your widgets are drawn on the screen. Four
geometry managers—pack,
grid, place, and

                     form — are covered here. Most of
the examples in the book use pack.


	
                  Chapter 3
               
	You can easily make effective use of fonts in your Perl/Tk
applications using Font objects. This chapter shows you how to
utilize Fonts and what options are available for changing them.
Several small applications are covered that demonstrate the use of
Fonts.


	
                  Chapter 4
               
	The Button widget is the first we cover, and we supply lots of
details. There are tons of code snippets and screen shots showing
different ways to manipulate the Button widget. Many of the options
we discuss are common among the other standard widgets. In addition
to the standard Button widget, we'll look at two derived
variants: the Checkbutton and Radiobutton widgets.


	
                  Chapter 5
               
	The Label widget is the simplest of all. It is usually used with an
Entry widget, which is why they are included in the same chapter.
Typically, the Entry widget accepts user input, and the Label
identifies the input. Perl/Tk has a special Tk::LabEntry widget that
we'll examine in detail.


	
                  Chapter 6
               
	Certain widgets in Perl/Tk can be scrolled, which means they can
contain more information than you can see on the screen. Scrollbars
are used to navigate the data inside these widgets. This chapter
tells you how Scrollbars communicate with each widget and how to
create and use them. It also illustrates the
Scrolled method, which automates Scrollbar
creation.


	
                  Chapter 7
               
	A Listbox widget can contain any sort of data, but it usually
contains a list of options from which the user can select. In this
chapter, you'll learn how to create a Listbox, fill it with
some items, and change the way the user selects items from the list.


	
                  Chapter 8
               
	The Text widget is a versatile widget you can use for many purposes
besides just displaying text. This chapter covers the different
things you can put inside a Text widget (such as text, images, or
other widgets) and how to get the best use out of them. The derived
TextUndo and ROText (Read-Only Text) widgets are also discussed.


	
                  Chapter 9
               
	A Canvas widget can display objects such as circles, rectangles,
text, images, and even other widgets. This chapter covers all the
options and methods available, and shows how to use them.


	
                  Chapter 10
               
	The Scale widget is great for giving the user a range of numbers from
which to select so that users can't type in numbers out of
range or type in letters accidentally. This chapter includes examples
of the Scale widget and covers all the methods available for setting
it up and using it.


	
                  Chapter 11
               
	The Frame and Toplevel widgets are used to organize your other
widgets on the screen to get the look you want. This chapter shows
how you can use Frames and Toplevels in coordination with a geometry
manager (covered in Chapter 2) to make your
windows look the way you want them to. We also look closely at the
MainWindow, which is a specialized Toplevel in disguise.


	
                  Chapter 12
               
	Once an application gets complex enough, you will need to put a Menu
in it. This chapter shows different ways to create menubars and
pulldown, popup, tearoff, and option menus, and how they can best be
used in an application. We also cover menu virtual events and briefly
examine pie menus.


	
                  Chapter 13
               
	There are many methods available for all widgets in Perl/Tk. We cover
them in this chapter and show you how to use them. The two most
important of these methods are configure and
cget.


	
                  Chapter 14
               
	Creating custom widgets is sometimes the only answer to a problem.
This chapter covers all the details, including the Tk class
hierarchy, and gives you several examples of composite and derived
mega widgets to examine. You will find details here that appear
nowhere else in the known universe. Featured widgets are Tk::Nil,
Tk::CanvasPlot, Tk::LabOptionmenu, Tk::LCD, Tk::NavListbox,
Tk::Thermometer, Tk::CollapsableFrame, and Tk::MacCopy.


	
                  Chapter 15
               
	This chapter explores the inner workings of Tk's event loop,
including timers, I/O, mouse and keyboard events, bindings, and
callbacks. Featured modules are Proc::Killfam, Tie::Watch, Tk::Trace,
Tk::bindDump, and Tk::waitVariableX. Featured widgets are
Tk::ExecuteCommand, Tk::MacProgressBar, and Tk::Splashscreen.


	
                  Chapter 16
               
	This chapter describes how to use the comand line and option database
to customize your Perl/Tk application.


	
                  Chapter 17
               
	This chapter covers the various image types and how to use them. We
examine Bitmaps, Pixmaps, Photos, and compound images, and touch on
tile, transparency, and animation issues. Featured widgets are
Tk::Animation, Tk::PhotoRotateSimple, Tk::Thumbnail, and
Tk::WinPhoto.


	
                  Chapter 18
               
	A detailed look at all the Tix widgets and ways to use them
effectively in Perl/Tk applications, including display items and
display styles.


	
                  Chapter 19
               
	With care, pipes and sockets can coexist with Tk's event loop.
This chapter develops two illustrative client/server programs.


	
                  Chapter 20
               
	Tk provides an unusual IPC mechanism that allows Tk programs to send
messages amongst themselves. This chapter describes Perl-Perl,
Tcl-Tcl, and Perl-Tcl intercommunications, and discusses security
considerations. We compute π with multiple processes and develop
a Perl plug-in for tclrobots so that Perl and
Tcl Robot Control Programs can do battle. Featured modules include
Tk::Receive and Tk::TclRobots.


	
                  Chapter 21
               
	This chapter shows how to write, debug, and package a Tk widget
written in C, using the Tk::Square widget as an example.


	
                  Chapter 22
               
	LWP is a Perl library for accessing the World Wide Web. This chapter
develops a web client, tkcomics, that displays
our favorite comic strips. It details various nonblocking mechanisms
for both Unix and Win32. Featured modules are LWP::Simple,
LWP::UserAgent, and Tie::Win32MemMap. We then describe the PerlPlus
Netscape plugin, which allows you to embed Perl in Netscape and run
client-side programs.


	
                  Chapter 23
               
	This chapter is a grab-bag of miscellaneous information and simple
widgets such as Adjuster, Balloon, BrowseEntry, ColorEditor, Dialog,
DialogBox, ErrorDialog, LabFrame, NoteBook, Pane, ProgressBar,
chooseColor, getOpenFile, getSaveFile, and messageBox.


	
                  Appendix A
               
	The Tk module doesn't come with the standard Perl distribution.
This appendix tells you where to download the latest release and
updates, and how to install them.


	
                  Appendix B
               
	This appendix lists all the options for every widget described in
this book.


	
                  Appendix C
               
	This appendix includes complete code listings of sample programs that
don't appear in the book proper.





Reading Order



This book was designed and written both for people new to Perl/Tk and
those who are familiar with it. How you approach the book depends on
which category best describes you.

Perl/Tk novices should probably start at the beginning. This book is
designed to lead you into topics by building a foundation of
knowledge. We'll start with a Hello World example in Chapter 1, and from there move into geometry management
and the standard widgets. Using Perl/Tk is not really that hard once
you understand the fundamentals of how it works.

Experienced programmers should feel free to skip around at will. We
recommend reading through Chapter 2 so you have a
complete understanding of how the geometry managers work. Then skip
around to the different sections in which you are interested. This
book has quite a few examples that will give you ideas on how to use
Perl/Tk in different ways.


Typographical Conventions



The following typographical conventions are used in this book:
	Italic
	Used for filenames, Unix command names, URLs, daemons, emphasis, and
the first use of terms where defined


	
                  Constant width
               
	Used for function and method names and their arguments, and to show
literal code in text


	
                  Constant width italic
               
	Used to identify replaceable values

	
                  Constant width bold
               
	Used to show default values in syntax lines and to indicate user input




We'd Like to Hear from You



Please address comments and questions concerning this book to the
publisher:

	O'Reilly & Associates, Inc.
	1005 Gravenstein Highway North
	Sebastopol, CA 95472
	(800) 998-9938 (in the United States or Canada)
	(707) 829-0515 (international or local)
	(707) 829-0104 (fax)

We have a web page for this book, where we list errata, examples, or
any additional information. You can access this page at:

	
               http://www.oreilly.com/catalog/mastperltk/
            

To comment or ask technical questions about this book, send email to:
	
               bookquestions@oreilly.com
            

For more information about our books, conferences, Resource Centers,
and the O'Reilly Network, see our web site at:

	
               http://www.oreilly.com

Chapter 1. Hello, Perl/Tk




         Tk is a
module that gives you the ability to create graphical interfaces with
Perl. Most Perl programs are written with command-line interfaces,
which can be cumbersome and intimidating to end users. Perl/Tk lets
you communicate with buttons, menus, dialog boxes, scrolled text
areas, and so on—all the features you need to develop simple or
sophisticated GUI applications.

Why use a graphical interface? In the course of your programming
experience, you've probably come across situations in which a
text-based interface was insufficient for your needs, if not
downright awkward. Certain applications can run with no input, but
others, such as installation scripts, require the user to feed
information to them constantly. They ask such questions as: Do you
want to install this file? Can I overwrite this DLL? Do you want to
create this directory? Do you want the help files?

A graphical user interface (GUI) adds a little flair and
professionalism to an application. Here are some examples of good
uses for a GUI:

	A mini web client that connects to a dictionary server

	An application that displays a map in a scrollable window

	A program that interfaces with a database and displays query results
in several widgets, with labels to describe the data


	A mail reader that interfaces with your inbox and can also send out
mail messages




A GUI can also be helpful when your boss just says "make it
easy to use!," which usually means either adding a wrapper
around a script or an interface that makes it easy for users to
understand the decisions they have to make.

But don't take this to mean that you should start adding GUIs
to all your Perl scripts. There are times when it would be overkill
to add a GUI to a script. If all you are doing is reading one file,
munging a bit with no user input, and generating another file, a GUI
would be silly and unnecessary. GUIs work best when you require a lot
of decisions and input from the user, such as in the installation
scenario mentioned earlier.

Perl/Tk Concepts




            
            
            
            Perl/Tk programs are written in an
object-oriented (OO) style, but you don't need previous Perl
object-oriented programming experience to code in Perl/Tk.
You'll pick it up easily enough after seeing the first few
examples. In a nutshell, Perl/Tk widgets (such
as Buttons and Listboxes) are objects that have
methods we invoke to control them. Besides
widgets, Perl/Tk has images, which are also
objects, and fonts, which can be objects or
simple strings.


            
            
            A
Perl/Tk program is composed of a hierarchy of
widgets. At the top of the hierarchy is the MainWindow, the parent
widget for all other widgets in the application. The MainWindow
widget acts as a container, within which we arrange child widgets
using a geometry manager. The widget hierarchy
is important for several reasons. Among other things, it's used
by geometry managers to control the screen layout and the menu system
to arrange menu items.


            
            Each
different widget belongs to a class. A
widget's class defines its initial appearance and behavior, but
individual widgets of the same class can be customized. As an
example, you might create two Buttons that have different textual
labels but are otherwise identical. Sometimes you'll read about
instantiating a widget. This is simply OO-speak
for creating a widget (a widget instance). The
class constructor is responsible for creating
widget instances.


            
            
            The
class also defines a widget's initial behavior by creating
bindings. A binding associates an
event such as a button press with a
callback, which is a subroutine that handles the
event. You can add additional bindings (indeed, even change and
remove them) to alter a widget's standard behavior. Callbacks
have several formats, but we mostly use simple references to Perl
subroutines.

You'll learn all about these topics as you continue reading.


Some Perl/Tk History



Perl/Tk has its roots in the X Window System and the Tcl language. So
let's take a detour into the pages of history, to give you an
idea of where Perl/Tk came from and how it got here.

The X Window System and Xlib




               
               The X Window System (known to its friends
as just "X" or "X11") was first released in
1987 as a graphical platform for Unix systems. Like most Unix
software, X applications are almost universally written in the C
language, using a library such as Xt, Motif, or (if you were really
unlucky or just really brave) the underlying library for X-based
applications, Xlib.


               Xlib has the
advantage that you can do anything, at the expense of dealing with
everything. For instance, here's one way to make a simple
pull-down menu using Xlib (which is one statement in Tk). First,
determine the dimensions of the longest menu item. For
argument's sake, assume the menu label string is in the C
variable menu_item. Subroutine
XTextExtents determines several metrics about
menu_item, such as its overall width in pixels in
the current font and its pixel height, computed by summing the
maximum ascent and descent (the number of pixels above and below the
baseline, respectively). After accounting for the number of menu
items, border widths, and including some slop for good luck, we
arrive at the dimensions of the menu window itself, and its relative
(x, y) position in the MainWindow.

  XTextExtents( font_info, menu_item, strlen( menu_item ),
    &direction, &ascent, &descent, &overall );
  menu_width = overall.width + 4;
  menu_pane_height = overall.ascent + overall.descent + 4;
  menu_height = menu_pane_height * menu_pane_count;
  x = window_width - menu_width - ( 2 * menu_border_width );
  y = 0;

               XCreateSimpleWindow draws the menu with the proper
border and background colors, although nothing appears on the display
because the window hasn't yet been mapped.

  theMenu = XCreateSimpleWindow( theDisplay, theWindow,
    x, y, menu_width, menu_height,
    menu_border_width, theBorderPixel,
    theBackgroundPixel );
But every menu item is itself a tiny window, so create them all, save
the structure pointers for later use, and select the events
they'll respond to. Notice that we haven't drawn the
actual text of the menu items. To do that, we need to define font and
graphic context items, then call XDrawImageString
to paint the characters (that's all done in initialization and
event handler code):

  for( i = 0; i < menu_pane_count; i++ ) {
    menu[i].menu_pane = XCreateSimpleWindow(
      theDisplay, theMenu, 0, menu_height/menu_pane_count*i,
      menu_width, menu_pane_height, menu_border_width = 1,
      theForegroundPixel, theBackgroundPixel );
    XSelectInput( theDisplay, menu[i].menu_pane, EVENT_MASK3 );
  }

  XMapSubWindows( theDisplay, theMenu );
The symbol EVENT_MASK3 enumerates the events
applicable to the menu item windows:

#define EVENT_MASK3 ButtonPresMask | ButtonReleaseMask |
  ExposureMask | EnterWindowMask | LeaveWindowMask
Now we must write the event handlers, including an Expose handler
that actually draws the windows, our own event loop, and even our own
event dispatching code, and on and on and on .... Whew!

Programming with higher-level libraries such as Motif or Xt is
somewhat more civilized, but it's no walk in the park either. A
significant stumbling block was that no matter what library you used,
X remained in the clutches of C programmers. C is a fine language,
but it kept X in the hands of the professionals—no hobbyists or
hackers need apply. If you needed to develop an in-house tool (e.g.,
for tracking bugs), many companies would balk at spending the time
and resources required for developing a C application, so you'd
end up with a clunky script with a command-line interface.

Something had to be done, and something was. 

The Coming of Tcl/Tk




               
               The Tool Command Language (Tcl) was
developed in 1987 by John K. Ousterhout of the University of
California at Berkeley. Ousterhout envisioned an embeddable,
extensible command language that many different applications could
reuse. Each application would inherit identical basic features such
as control structures, scalar variables and arrays, and built-in
procedures. In turn, an application would add its unique commands,
each of which had the same "feel" as any other Tcl
command.[1]
            

               But
Ousterhout needed to prove his ideas, as much to himself as to
others. Since he was interested in GUIs, he devised a toolkit of
graphical components and tied them together using Tcl. He reasoned
this approach would be more cost effective than writing C language
code, even using a toolkit like Motif. His hunch was proven correct,
bringing us to his next accomplishment: his graphical toolkit called
Tk, from which Perl/Tk is derived.

In early 1991, Ousterhout released Tk Version 1.0, the graphical
extension to the Tcl scripting language. Tk's high-level widget
set (which ultimately uses Xlib as its drawing package) was an
immediate hit. In the years following, thousands of Tcl/Tk
applications were written and Tk was ported to languages such as
Eiffel, Modula-3, Prolog, Python, Scheme, and more.


               
               
               By November of 1993, Tcl/Tk was at
Version 3.4 and, believe it or not, folks were busy pasting Tcl/Tk
GUIs on top of their Perl programs. If only we Perlers had known that
help was on its way, for that very same month Malcolm Beattie of
Oxford University began his TkPerl
project.[2]
            

The Evolution of Perl/Tk




               Malcolm's goal was a pure Perl 5,
object-oriented interface to Tk without any dependence on Tcl, which
meant converting Tcl code to Perl and writing XSUBs so Perl could
call Tk C library routines. The marriage of Tk and Perl was
complicated further because in those days Tcl/Tk C subroutines passed
simple strings back and forth, which didn't fit well with
Perl's model of native data types. Nonetheless, by the summer
of 1994, TkPerl was available in alpha form for general use, sans the
Text widget and a handful of lesser-used commands and bindings.

Here's an early TkPerl "Hello World" program:
use Tk;
$mw = tkinit;
$b = Button::new($mw, -text => 'Hello World');
$b->configure(-method => sub {exit});
tkpack($b);
tkmainloop;
If this looks odd to you, remember it was the state of affairs nearly
eight years ago, when TkPerl was alpha, and Perl's
object-oriented features were still beta. Notice that
pack hadn't yet been turned into a widget
method and was renamed tkpack so it didn't
conflict with Perl's built-in function by the same name. Yet it
worked, and it let us use Tk from Perl, which, after all, was the
goal.


               
               
               
               
               Around this time, another chap from the
United Kingdom, Nick Ing-Simmons (then of Texas Instruments), began
using TkPerl in earnest. He and Malcolm collaborated for a time, and
they mutually agreed that Nick would continue development. From this
came nTk, or "new Tk," or possibly "Nick's
Tk," and thus began the evolution of Perl's Tk
programming interface to what it is today. In May 1995, there was
another name change, and Nick's package became known as pTk,
for "pure Tk," or "portable Tk," or
"Perl/Tk." Throughout the years, user-contributed widgets
and Ioi Kim Lam's Tk Interface Extension (Tix) widgets found
their way into the distribution. These Tks were all based on Tcl/Tk
Version 4.x, a version for Unix only.

Remember the Xlib code we showed? Figure 1-1 shows
a simple Perl/Tk window.

[image: Hello, Perl/Tk]

Figure 1-1. Hello, Perl/Tk

This window was created with the following Perl one-liner:
perl -MTk -e 'MainWindow->new->Label(-text => "Hello, Perl/Tk")->pack; MainLoop'
This example highlights just how far we've come. 

Perl/Tk Meets Win32




               In the
meantime, the Tcl/Tk team members weren't resting on their
well-deserved laurels. By the summer of 1995, Tcl/Tk 8.x[3] was running on Unix, MacOS classic,
and Win32 operating systems, with a look and feel appropriate for
each environment. It would take two more years for Perl/Tk to catch
up.[4]
            

               By the
summer of 1997, Gurusamy Sarathy, a well-known and respected Perl
porter, had produced a binary distribution of Perl 5.004 and Tk
40x.000 (and other useful modules) specifically for Window 95 and
Windows NT. The growth of Perl/Tk took off exponentially. The only
major complaint was that a Perl/Tk GUI looked too Motif-like, due to
its Unix roots.


               Nick, with his
prodigious programming abilities, wasn't idling either, for in
early 1998 he'd merged all of Perl/Tk, Tix, Jan Nijtmans'
image package, and Tcl/Tk 8.0, thus creating the basis for the
current Perl/Tk 800.000 series.

Simultaneously, Sarathy was heading the effort to combine the
disparate Perl ports into a unified Unix and Win32 distribution,
commonly called Oneperl. Finally, in the summer of 1998, we had one
Perl, 5.005, for Unix and Win32. ActiveState Corporation distributes
this unified Perl in binary form, used on most Win32 systems these
days.

Around July 1998, Nick produced a unified Unix and Win32 Perl/Tk
distribution, and placed a binary version of Perl/Tk 800.010 in
ActiveState's PPM repository. The result is that Win32 users
can simply download binary installation packages of Perl and Tk. For
most Unix users, using a simple idiom, you compile Perl and Tk
yourself. If you're really lucky, you can search the Web and
find a binary distribution of Perl/Tk for your particular flavor of
Unix.

Running Perl/Tk Programs on Win32
There are no differences between writing Perl/Tk applications on Unix
or Windows machines. You can use any simple text editor on either
system. However, there can be a small difference in the way you run
them.

The lowest common denominator is to manually invoke Perl and specify
the Perl/Tk program filename on the command line, like this:

perl myprog.pl
A Unix user commonly gives his program execute permission and ensures
that the first line of the program is a valid "bang"
line. This allows the user to invoke Perl/Tk programs by name. When
invoking a program by name, the Unix command processor, called the
shell program, inspects the file's first line and, if it begins
with the characters #!, treats the remainder of
the line as the command to execute, possibly with arguments. The
! character is the bang. A
line of this form invokes Perl with the -w switch:

#!/usr/local/bin/perl -w
Conceptually, Win32 users do the same thing by associating the
extension .pl with the Perl interpreter, so they
can use Explorer and double-click the script to execute it. See Appendix A for more details.





[1] Years earlier, Control Data Corporation
carried the concept of an embeddable, extensible command language to
a logical conclusion with its operating system, NOS/VE. The command
language was called System Command Language (SCL) and as the name
implies it was used by the entire operating system, from utilities,
compilers, and debuggers, to terminal servers and end-user scripts
and applications. Any application could embed any other
application—automatically, courtesy of the operating
system—without any work from the user. The
symmetry and consistency between applications was most
amazing.

[2] While TkPerl is no longer
available, Malcolm has two CPAN modules that allow you to use Tcl/Tk
commands from a Perl script. Of course, they rely on Tcl/Tk
libraries, so you need Tcl/Tk installed.

[3] There were no Tk 5.x, 6.x, or 7.x versions. Tk jumped from a
4.x version number to 8.x to match the corresponding Tcl version.
This made it easier to know which version of Tk matched which Tcl
installation.

[4] On Windows, that is; Perl/Tk is still unavailable
on Mac OS at the time of this writing.



Getting Started with Perl/Tk



Enough history. The remainder of this chapter is about the basics of
using Perl/Tk, including how to create simple widgets and display
them, a quick introduction to event-based programming, and an
obligatory "Hello World" example. Before we continue,
let's make sure you have everything installed properly.

Do You Need To Install Anything?




               Since
the Tk extension to Perl doesn't come with the standard Perl
distribution, the first thing you should do is make sure you have a
working Perl/Tk distribution.


               
               
               
               
               
               
               Whether you're running Unix or
Win32, the perl program should be in your path.
Type the following at a command prompt to make the determination:

% 
               perl -v
If you receive a "command not found" error message, see
Appendix A and install Perl. If
perl is found, you'll see output similar
to this:

This is perl, v5.6.0 built for i686-linux

Copyright 1987-2000, Larry Wall

Perl may be copied only under the terms of either the Artistic License or the
GNU General Public License, which may be found in the Perl 5.0 source kit.

Complete documentation for Perl, including FAQ lists, should be found on
this system using `man perl' or `perldoc perl'.  If you have access to the
Internet, point your browser at http://www.perl.com/, the Perl Home Page.
In particular, note the version number in the first line; anything
earlier than 5.6.0 may not produce the results depicted in this book.
Perl 5.005_03 may work, but nothing earlier
will, guaranteed.

Now determine if the Tk module is available by using this command:
% perl -e "use Tk"
If you don't get the following error, you're ready to go:
Can't locate Tk.pm in @INC (@INC contains: C:\PERL\lib\site ...
Once again, to install Tk refer to Appendix A.

               Assuming that
Perl/Tk is up and running, you can determine its version with this
command:

% perl -MTk -e 'print "$Tk::VERSION\n"'
800.022
Ideally, you want Version 800.022 or higher.

               
               
               The best way to verify that all is well is
to run the widget demonstration program.
widget should already be in your path, so just
invoke the command by typing widget at a
command prompt. This program demonstrates most of the Perl/Tk widget
set and lets you examine the Perl/Tk code, modify it, and rerun
individual demonstrations. Clicking on About shows you the installed
Perl and Tk versions.

Perl/Tk Versions
This book is based on the stable releases of Perl 5.6.0 and Tk
800.022. At the time of this writing, Perl 5.6.1 has been released,
yet Tk continues to work as before, and all examples are known to
work. Down the pipe is a major release of Tk, based on Tcl/Tk Version
8.3. Nick has seeded Tk 803.023 to a few Perl/Tk hackers, but the
code is still far from prime-time ready.

When Perl/Tk 803.xxx becomes available, expect it to be thread-safe
and Unicode (UTF-8) aware, but beware that if your application uses
high-bit ISO-8859-1 characters, it will most likely break.



Creating Widgets




               
               All widgets in Perl/Tk programs are
created in the same basic fashion, with a few exceptions. Each widget
must have a parent widget to watch over it as it is created and keep
track of it while it exists in the application. When you create an
application, you'll have a central window that will contain
other widgets. Usually that window will be the parent of all the
widgets inside it and of any other windows you create in your
application. You are creating an order to the widgets so that the
communication between child and parent widgets can happen
automatically without any intervention from you once you set it all
up.


               Assuming
that the $parent widget already exists, the
generic usage when you create widget Widgettype is
as follows:

$child = $parent->Widgettype( [ -option => value, . . . ] );

               
               Note
that the variables that store the widgets are scalars. (Actually,
they are references to widget objects, but you don't need to
know that right now.) If you aren't familiar with
object-oriented syntax in Perl, using the ->
between $parent and Widgettype
invokes the method Widgettype from the
$parent object. It makes the
$parent related to the child
$child. As you might guess, the
$parent becomes the parent of the widget being
created. A parent can have many children, but a child can have only
one parent.


Specifying Options




               
               
               
               
               When you invoke the
Widgettype method, you usually specify
configuration parameters to set up the widget and the interactions
within the application. The configuration parameters will occur in
pairs: an option (such as -text,
-state, or -variable) and its
associated value. Each option starts with a dash, but that's
only by convention; the options are just strings used to indicate how
to interpret their associated values.


               
               
               Usually, it is not necessary to put
quotation marks around option names because Perl is smart enough to
recognize them as strings. However, if you are using the
-w switch, Perl may complain about an option
that it thinks is not text. You can stick quotes around all your
options all the time to avoid this, but it shouldn't be
necessary. The option names are all lowercase, except in a few rare
cases that we'll note as we cover them.


               
               
               Options
are specified in list form:

(-option => value, -option => value, -option => value)

               
               
               If you've never seen
=> in Perl before, don't be thrown by it.
It's just a different way of saying "comma," except
that the => operator auto-quotes the word to
its left, eliminating possible ambiguities. For instance, the
following code works properly because the auto-quoting resolves
-text as a string:

sub text {}
$mw->Label(-text => 123);
With the comma syntax, however, -text resolves to
-&text( ):

$mw->Label(-text, 123);
With this in mind, you can still use just the commas and not the
=> notation, such as:

(-option, value, -option, value, -option, value)
However, it's much harder to tell which are the option/value
pairs. Consider the following syntactically equal statements (each of
which create a Button widget that is 10 by 10 pixels, displays the
word "Exit," and performs the action of quitting the
application when pressed):

$bttn = $parent->Button(-text, "Exit", -command, sub { exit }, -width, 10, -height, 10);

$bttn = $parent->Button(-text => "Exit", -command => sub { exit }, -width => 10, -height => 10);
In the second line, it is much more obvious which arguments are
paired together. The option must be directly before the value
associated with it: -text is paired with
"Exit", -command has the value
sub { exit
               }, and -width and
-height both have values of 10.


               
               Another favorite option/value
specification syntax uses Perl's qw
operator, which treats its arguments as a list of strings:

$bttn = $parent->Button(qw/-text Exit -width 10 -height 10 -command/ => sub { exit }/);

               
               
               
               
               
               
               
               This style is more reminiscent of
Tcl's look, with whitespace-separated tokens. You tend to type
fewer characters too. The string delimiter is often (
) or {}, but // is
most popular since it doesn't require a shift. Note that
qw splits on simple words, so that option values
can be only simple words, not multiword quoted strings, code
references, and so on. That's why we moved the
-command option to the end of the
qw string.


Toplevel, MainWindow, and Frame Widgets



Time for another detour. In the next few chapters we'll be
using widgets in our examples that we might not have covered yet. We
trust that you'll figure out what most of them mean from the
context in which they are presented, but a few require a short
introduction.


               
               
               MainWindow
and Toplevel are the windows (or widgets—we often interchange
the terms) that contain other widgets. MainWindow is a special
version of a Toplevel widget, in that the MainWindow is the first
window you create in your application.


               
               The other type of widget you need
to know about is a Frame widget. A Frame is a container that can also
contain other widgets. It is usually invisible and is used just to
arrange the widgets as desired.

Of course, there's more to it, but that's enough to know
for now. For more information, see Chapter 11.


Displaying a Widget




               
               
               
               Creating a
widget isn't the same as displaying it in Perl/Tk. You need to
use two separate commands to create a widget and display it, although
sometimes they are combined into the same line and look like a single
command. In the examples so far, we've used the
Button method to create the Button, but nothing is
displayed by using that method alone. Instead you have to use a
geometry manager to cause the widget to be displayed in its parent
widget or in another widget. The most commonly used geometry manager
is pack. To use it, you simply call the
pack method on the widget object, as follows:

$widget->pack(  );
For example:
$button->pack(  );
The arguments you can send to the pack method are
covered in Chapter 2.

It is not necessary to invoke the pack method on a
separate line. ->pack can be added to the
creation of the widget:

$parent->Button(-text => "Bye!", -command => sub { exit })->pack(  );

               
               
               
               The other geometry managers available
are grid, form, and
place. All four behave differently; use what works
best for your application. Again, look for information on the
geometry managers in Chapter 2.


The Event Loop




               
               When
programming an application that uses a graphical interface rather
than a textual interface, you need to rethink the way you approach
the flow of the application. In a text-based application, you can
read from standard input (STDIN), use command-line options, read
files, or prompt the user for specific information. The keyboard is
your main avenue of input from the user. In a GUI, input comes not
only from those places but also from the mouse and the window
manager.[5] Although this extra input allows more
flexibility in our applications, it also makes our programming job
more difficult. As long as we tell it what to do, Perl/Tk helps us
handle all that extra input gracefully.


               Input
in a GUI is defined by events. Events are typically different
combinations of using the keyboard and mouse at the same, or
different, times. If the user pushes the left mouse button on Button
"B," that is one type of event. Pushing the right mouse
button on Button "C" is another event. Typing the letter
"a" is another event. Holding down the Control key and
clicking with the middle mouse button is yet another event. Events
can also come from input and output operations or be generated
virtually under program control. For an in-depth examination of the
Tk event loop, see Chapter 15.


               Events are
processed during an event loop. The event
loop, as its name implies, handles events during a loop. It
determines what subroutines to call based on what type of event has
happened. Here is a pseudocode event loop:

while (1) {
  get_event_info

  if event is left-mouse-click call process_left_mouse_click
  else if event is right-mouse-click call process_right_mouse_click
  else if event is keyboard-input call type_it
  else handle events for redrawing, resizing etc
}
This is obviously a simplistic approach to an event loop, yet it
shows the basic idea. The event loop is a weeding-out process to
determine what type of input has been given to the application. For
example, the subroutine process_left_mouse_click
might determine where the pointer was when the mouseclick occurred
and then call other subroutines based on that information.


               In Perl/Tk, the event loop is initiated
by calling a routine called MainLoop. Anything
prior to this statement is just setting up the interface. Any code
after this call will not execute until after the GUI has exited using
$mw->destroy.[6]
            
If we forget to include the MainLoop statement,
the program will think about things for a while and then go right
back to the command prompt. None of the windows, Buttons, or widgets
will be drawn at all. The first things that occur after calling
MainLoop are the interface is drawn and the event
loop is started.

Before we get too much further into the event loop and what it does
(and what you need to do so it works properly), let's look at a
working example program, Hello World. (You were expecting something
else?)




[5] For example, a "close" directive
from a window manager such as mwm or MS
Windows.

[6] Throughout the
book, we use $mw to indicate the variable that
refers to the MainWindow created at the beginning of the
application.



Hello World Example




            Every
programming language goes through the Hello World example, which is a
complete program that prints a string (typically "Hello
World") and exits. "Hello World" may get its share
of ridicule, but it's a remarkably effective tool that shows
readers how to write and execute a working program while
they're still in the first chapter of the book. In our Hello
World example, we'll have the title of our window say
"Hello World" and create a Button that will dismiss the
application:

#!/usr/bin/perl
use Tk;
my $mw = MainWindow->new;
$mw->title("Hello World");
$mw->Button(-text => "Done", -command => sub { exit })->pack;
MainLoop;
Despite being only six lines long, there is quite a bit going on in
our little program. The first line, as any Perl programmer knows,
invokes Perl.[7] The second line tells Perl to use
the Tk module.

The third line:
my $mw = MainWindow->new;
is how we create a window. The window will have the same basic window
manager decorations as all your other windows.


            The title of
our window is changed using the title
method. If we hadn't used this method, the
text across the top of the window would be the same as the name of
the file containing the code, excluding any extension. For instance,
if the code were stored in a file named
hello_world, the string
"Hello_world" would appear across the title bar of the
application (Tk automatically capitalizes the first character for
you). Using the title method is not required, but
it makes the application look more polished.

Any string we put as an argument becomes the title. If we wanted the
title to be "Hey! Look at my great program!," this would
be the place. This is akin to using the -title
option when starting any standard X Windows application. We cover
more methods for a MainWindow object later in Chapter 11.

The next line creates a Button widget, sets basic properties, and
packs the widget. (See Chapter 4 for all available
configuration options for Button.)

The Button is set to display the text "Done" and to
perform the Perl command exit when pushed.
Finally, the last item of concern is the MainLoop
command. This starts the event loop in motion, and from then on the
application will do only what we have told it to do: if the user
clicks on the Button, the application will exit. Anything else the
user does—minimizing, resizing, changing to other
applications—will be processed by the window manager and
ignored by our application. See Figure 1-2 for a
picture of the Hello World window.

[image: Hello World window]

Figure 1-2. Hello World window

exit Versus destroy




               In most of
the examples in this book, you will see sub
               { exit; } (
or its equivalent, \&exit) used to quit the
Perl/Tk application. This works fine as long as you have done a
use Tk; in the same file. Perl/Tk defines its own
exit routine, which does some cleanup and various
other things that are important to Tk. The program is then
unconditionally terminated, and control returns to the operating
system.


               Another way to quit the Tk portion of the
application is to call $mw->destroy, which
destroys the MainWindow and returns to the code listed after
MainLoop. This allows your program to do post-GUI
processing before exiting.




[7] On Unix, that is. In Win32 you have to
type perl hello.pl to invoke the program or
twiddle with Explorer to call up the perl
executable when .pl files are
double-clicked.



Unsolicited Advice



Before we end this chapter, and you become engrossed in the details
of Perl/Tk, we'd like to give you some suggestions on
programming style and window design. Bear with us, this won't
take long and might save you a lot of time in the future.

Programming Style




               
               The
code in a Perl/Tk script can get quite cumbersome and clunky because
of all the option/value pairs used to define and configure each
widget. There are several ways to format the code to deal with
readability (and in some cases, "edit-ability"). Most
just involve adding extra spaces or tabs to line up different
portions of code. Once you get used to seeing the code, it
won't seem quite so mysterious and unwieldy.

One coding style places each option/value pair on a separate line:
$bttn = $parent->Button(-text => "my text",
                        -command => sub { exit }, 
                        -width => 10, 
                        -height => 10);

               With this style, it is extremely
obvious what the pairs are and what value is associated with which
option. (You could also go to the extreme of aligning each
=> to make nice columns, depending on how much
time you have to press the spacebar.) Some people like to start the
option/value pairs on the next line and put the ending
); on its own separate line, after the last
option/value pair, which retains the comma for formatting ease:

$bttn = $parent->Button(
   -text => "Exit", 
   -command => sub { exit }, 
   -width => 10, 
   -height => 10,
);

               This
makes the code easier to edit; an option/value pair can be added or
deleted on each line without having to mess with parentheses,
semicolons, or commas. It also keeps the next lines closer to the
left side of the page, so if you have several indentation levels, you
don't end up with code quite so deeply nested to the right.


               
               In either case, Emacs users
may find the functionality of cperl-mode.el
handy. This is an Emacs initialization file that adds color
highlighting and special formatting that makes editing Perl code more
efficient. You can find the file in the standard Perl distribution,
in the perl-5.6.0/emacs directory.

Sometimes if there are only one or two option/value pairs, it makes
sense to leave them all on the same line and conserve a little bit of
space:

$bttn = $parent->Button(-text => "my text", -command => sub { exit });
You'll eventually come up with a style that works for the way
you read and edit code. Whichever way you choose, try to be
consistent throughout your scripts in case someone else takes over
the maintenance of your code (it might even be you, a year or more
down the road).


Naming Conventions for Widget Types



Sometimes your programs run away from you, getting so large and
unwieldy that it becomes hard to remember what a particular variable
was pointing to. If there are over 10 Buttons in a program, we would
be hard-pressed to figure out which Button was
$button3 without digging through a bunch of code.

So we've established a naming convention for our code. You
don't need to use our convention, but it'll hopefully
inspire you to come up with your own. Otherwise, we hope you have a
really good memory.


               
               For Buttons, you might use
_b, _bttn, or
Button as a type of qualifier to the variable
name. For instance, you could name the Button in the Hello World
example $done_b,
$done_bttn,or$doneButton.


               The topmost widget in an
application's widget hierarchy is the MainWindow. Throughout
this book, we use the variable $mw to represent
this widget. You will see other programs use
$main, $top, or
$mainwindow as well.


               Table 1-1 lists widget types and suggested naming
conventions for them. Replace "blah" with a sensible
description of the widget's purpose (e.g.,
exit). If you use this convention, you'll
always know what type of widget you're working with.

Table 1-1. Naming conventions by widget type 
	
                           Widget type

                        	
                           Suggested name

                        	
                           Examples

                        
	
                           Button

                        	
                           
                              $blah_b, $blah_bttn,
or $blahButton
                           

                        	
                           
                              $exit_b, $apply_b,
$newButton
                           

                        
	
                           Canvas

                        	
                           
                              $blah_canvas or $blahCanvas
                           

                        	
                           
                              $main_canvas, $tinyCanvas
                           

                        
	
                           Checkbutton

                        	
                           
                              $blah_cb or $blahCheckbutton
                           

                        	
                           
                              $uppercase_cb
                              
                              
                              
                              , $lowercaseCheckbutton
                           

                        
	
                           Entry

                        	
                           
                              $blah_e or $blahEntry
                           

                        	
                           
                              $name_e, $addressEntry
                           

                        
	
                           Frame

                        	
                           
                              $blah_f or $blahFrame
                           

                        	
                           
                              $main_f, $left_f,
$canvasFrame
                           

                        
	
                           Label

                        	
                           
                              $blah_l or $blahLabel
                           

                        	
                           
                              $name_l, $addressLabel
                           

                        
	
                           Listbox

                        	
                           
                              $blah_lb or $blahListbox
                           

                        	
                           
                              $teams_lb, $teamsListbox
                           

                        
	
                           Menu

                        	
                           
                              $blah_m or $blahMenu
                           

                        	
                           
                              $file_m, $edit_m,
$helpMenu
                           

                        
	
                           Radiobutton

                        	
                           
                              $blah_rb or $blahRadiobutton
                           

                        	
                           
                              $blue_rb, $grey_rb,
$redRadiobutton
                           

                        
	
                           Scale

                        	
                           
                              $blah_scale or $blahScale
                           

                        	
                           
                              $age_scale, $incomeScale
                           

                        
	
                           Scrollbar

                        	
                           
                              $blah_scroll, $blah_sbar, or
$blahScroll
                           

                        	
                           
                              $x_scroll, $yScroll
                           

                        
	
                           Text

                        	
                           
                              $blah_t or $blahText
                           

                        	
                           
                              $file_text, $commentText
                           

                        
	
                           Toplevel

                        	
                           
                              $blah_w or $blahWindow
                           

                        	
                           
                              $main_w, $fileopenWindow
                           

                        




Designing Your Windows 



Before you decide what events to handle, it is worthwhile to spend
some time sketching out a few windows on paper and deciding what
should happen (from the user's perspective) when you click a
button or invoke a menu item.

A GUI often makes the application look much more polished and
purposeful than a command-line interface does, but it's easy to
go overboard with a GUI and end up with something ugly, clunky, and
impossible to navigate. So here are some things to consider when
deciding how the GUI should look:

	Every widget should have a purpose that is intuitive and informative.

	Think about the way a user will use an application and design
accordingly.


	Don't try to cram everything your application does into one
window.


	Don't always try to separate everything into different windows.
Sometimes the application is so simple that one window is all you
need.


	Colors are great, but there are a lot of color-blind people out
there. The same applies to fonts: many folks cannot read very small
fonts. If you insist on using color and particular fonts, allow them
to be customized via the resource database, through a file, or
through the application itself.


	Some widgets do their jobs better than others. Use the proper widget
for the job.






Debugging and PrototypingPerl/Tk Programs




            
            
            
            
            Debugging a Perl/Tk program need not
be different from debugging a nongraphical program; you can always
sprinkle warn statements throughout the code to
track progress and display intermediate results. We suggest using
warn rather than print for
three reasons: it adds the newline to the message automatically; the
output includes the line number of the warn
statement; and the output goes to STDERR, which is not normally
buffered, thus the output appears immediately. Furthermore, you type
fewer characters.

You normally run programs by typing the program name at the command
prompt:

% hello_world
or: 
C:\>perl hello_world

            
            
            When you invoke the program this way, any
printed output goes to that terminal window. If you don't put a
\n on the end of the string to be printed, you
won't see the information actually printed until you quit the
program. You may have to unbuffer a file handle by setting the
special Perl variable $|. If you use
warn rather than print, these
drawbacks are eliminated.


            
            
            
            If that old-fashioned way isn't to
your liking, perhaps the slightly newer old-fashioned way of using
the standard Perl debugger is. The debugger has built-in Tk support,
though you must use the O command and enable it by
setting the variable tkRunning:

[bug@Pandy atk]$ perl -de 0
Default die handler restored.

Loading DB routines from perl5db.pl version 1.07
Editor support available.

Enter h or `h h' for help, or `man perldebug' for more help.

main::(-e:1):   0
  DB<1> O tkRunning
           tkRunning = '1'
  DB<2> use Tk    
  DB<3> $mw = MainWindow->new
  DB<4> $b = $mw->Button(-text => 'Beep', -command => sub{$mw->bell})
  DB<5> $b->pack
  DB<6> x $b
0  Tk::Button=HASH(0x82ed434)
   '_TkValue_' => '.button'
  DB<7> q
As you see, we can not only print debug information, but also do
simple prototyping.


            An even better
environment for this sort of activity is the program
ptksh. It's part of a standard Perl/Tk
installation and, as its name suggests, it's a Perl/Tk shell
that allows us to interactively enter and test Perl and Tk commands.
Figure 1-3 shows a sample ptksh
session.

[image: ptksh session]

Figure 1-3. ptksh session


            
            If you're really into graphical
debugging, treat yourself to the CPAN module
Devel::ptkdb, an excellent, sophisticated
Perl/Tk debugger. Simply invoke Perl with a -d
argument such as this:

[bug@Pandy atk]$ perl -d:ptkdb group

            Figure 1-4 shows a ptkdb
session.

[image: ptkdb session]

Figure 1-4. ptkdb session


Chapter 2. Geometry Management




         To
display widgets on the screen, they must be passed to a
geometry manager. The geometry manager
controls the position and size of the widgets in the display window.
Several geometry managers are available with Perl/Tk:
pack, place,
grid, and form.


         All the geometry managers are invoked as
methods on the widget, but they all have their own methodologies and
arguments to change where and how the widgets are put on the screen:

$widget1->pack(); $widget2->place(); $widget3->grid(); $widget4->form(  );

         
         When you organize the widgets in
your window, it is often necessary to separate groups of widgets to
get a certain look and feel. For instance, when you use
pack, it is difficult to have widgets stacked both
horizontally and vertically without grouping them in some fashion. We
use a Frame widget or another window (a Toplevel widget) to group
widgets inside a window.


         We create our first window
by calling MainWindow. The MainWindow is a special
form of a Toplevel widget. For more detailed information on how to
create/configure Frame and Toplevel widgets, see Chapter 11.


         With the
exception of place, differences between the
geometry managers make it difficult (not entirely impossible, but
definitely not recommended) to use more than one geometry manager
within the same area. In
$mw, we can display many types of widgets, but if
we start using pack, we should continue to use
pack on all the widgets contained directly in
$mw. Don't switch to grid
in the middle, because the two geometry managers will get into a
race 
         condition: one will
create its layout, which affects the geometry calculations of the
other, which affects the layout of the first, causing it to recompute
its geometries, ad infinitum. However, let's assume our
MainWindow contains a Frame, which in turn contains other widgets. We
could use pack to pack the Frame inside the
MainWindow and then we could use grid to manage
the widgets inside the Frame. See Figure 2-1.

[image: Frame within a window that uses a different geometry manager]

Figure 2-1. Frame within a window that uses a different geometry manager


         Although the different geometry managers
have their own strengths and weaknesses, pack is
the most commonly used, so we'll discuss it first and in the
most detail. The grid geometry manager has been
improved greatly with the release of Tk 8.0 and subsequent porting to
Perl. The place geometry manager is the most
tedious to use, because you have to determine exact coordinates
(relative or absolute) for every single widget. Finally, the
form geometry manager is like a combination of
pack and place.

The pack Geometry Manager




            Remember
when you were a child and you had those wooden puzzles to put
together? Each piece in the puzzle had exactly one place where it
could go and there weren't any overlaps allowed between pieces.


            With the pack geometry
manager, our windows are similar to the wooden puzzle, because
widgets cannot overlap or cover each other, partially or completely
(see Figure 2-2). If a Button is packed in a
certain space on the window, the next Button (or any widget) will
have to fit around the already packed Button. Luckily, our windows
will be dealing only with rectangular shapes instead of funny-shaped
puzzle pieces.

[image: Overlap error]

Figure 2-2. Overlap error

The order in which you pack your widgets is very important because it
directly affects what you see on the screen. Each Frame or Toplevel
maintains a list of items that are displayed within it. This list has
an order to it: if widget A is packed before widget B, then widget A
will get preference if space becomes scarce. This will become clear
as we go through some examples. You will often get a different look
to your window just by packing the widgets in a different order.

If you don't care what the window looks like and how the
widgets are put in it, you can use pack with no
arguments and skip the rest of this chapter. Here it is again:

$widget->pack( );

            To make your
window look nicer and more manageable (and user friendly), there are
arguments that can be sent to the pack method that
will change the way the widgets and the window look. As with anything
in Perl/Tk, the arguments are arranged in key/value pairs. So the
more sophisticated usage would be:

$widget->pack( [ option => value, ... ] );
Here is the code to create a window that doesn't use any
pack options. We haven't covered all the
widgets used in this example, but hang in there; it's pretty
simple.

#!/usr/bin/perl -w
use Tk;

my $mw = MainWindow->new;
$mw->title("Bad Window");
$mw->Label(-text => "This is an example of a window that looks bad\nwhen you don't
 send any options to pack")->pack;

$mw->Checkbutton(-text => "I like it!")->pack;
$mw->Checkbutton(-text => "I hate it!")->pack;
$mw->Checkbutton(-text => "I don't care")->pack;
$mw->Button(-text => "Exit",
            -command => sub { exit })->pack;
MainLoop;

            Figure 2-3 shows the resulting window.  
[image: Window with widgets managed by pack]

Figure 2-3. Window with widgets managed by pack

We can alter the preceding code and add some options to the
pack calls that will make our window look much
nicer:

#!/usr/bin/perl -w
use Tk;

my $mw = MainWindow->new;
$mw->title("Good Window");
$mw->Label(-text => "This window looks much more organized, and less haphazard\n" .
    "because we used some options to make it look nice")->pack;

$mw->Button(-text => "Exit",
            -command => sub { exit })->pack(-side => 'bottom', 
                                            -expand => 1,
                                            -fill => 'x');
$mw->Checkbutton(-text => "I like it!")->pack(-side => 'left',
                                              -expand => 1);
$mw->Checkbutton(-text => "I hate it!")->pack(-side => 'left', 
                                              -expand => 1);
$mw->Checkbutton(-text => "I don't care")->pack(-side => 'left', 
                                                -expand => 1);
MainLoop;

            Figure 2-4 shows the much more organized window.
[image: Window with widgets managed by pack using some options]

Figure 2-4. Window with widgets managed by pack using some options


            
            
            
            Using
pack allows you to control the:

	Position in the window relative to the window or Frame edges

	Size of widgets, relative to other widgets or absolute

	Spacing between widgets

	Position in the window's or Frame's widget list



The options, values, and defaults are listed and discussed in the
following section.

Options for pack




               This list shows all the options available
when you call pack (the default values are shown
in bold):

	
                     
                     
                     
                     
                     
                     
                     
                     
                     
                     
                     
                     
                     -side => 'left' | 'right' | '
                     top
                     ' | 'bottom
                  
	
                        
                        
                        
                        
                        
                        
                        
                        
                        Puts
the widget against the specified side of the window or Frame


	
                     
                     
                     
                     
                     
                     -fill => '
                     none
                     ' | 'x' | 'y'| 'both'
                  
	
                        
                        
                        Causes the widget to fill the allocation
rectangle in the specified direction


	
                     
                     
                     
                     
                        0
                     
                  
	
                        
                        
                        Causes the
allocation rectangle to fill the remaining space available in the
window or Frame


	
                     
                     
                     -anchor => 'n' | 'ne' | 'e' | 'se' | 's' | 'sw' | 'w' | 'nw' | '
                     center
                     '
                  
	
                        
                        Anchors the widget inside the
allocation rectangle


	
                     -after => $otherwidget
                  
	
                        Puts $widget after
$otherwidget in packing order


	
                     -before => $otherwidget
                  
	
                        Puts $widget before
$otherwidget in packing order


	
                     -in => $otherwindow
                  
	
                        Packs $widget inside of
$otherwindow rather than the parent of
$widget, which is the default


	
                     
                     -ipadx => 
                     amount
                  
	
                        
                        Increases the size of the widget
horizontally by amount
                     

	
                     -ipady => 
                     amount
                  
	
                        
                        
                        Increases the size of the widget
vertically by amount
                     

	
                     -padx => 
                     amount
                  
	
                        
                        Places
padding on the left and right of the widget


	
                     
                     -pady => 
                     amount
                  
	
                        
                        
                        
                        
                        
                        
                        
                        Places padding on the top and bottom of
the widget





Positioning Widgets




               
               
               
               Each window (or Frame) has four
sides to it: top, bottom, left, and right. The packer uses these
sides as points of reference for widgets. By default,
pack places the widgets against the top of the
Toplevel or Frame.


               You can control
the side a widget is placed against with the -side
option:

-side =>  'left' | 'right' | 'top' | 'bottom'
For example, if we would like our Button against the left edge of the
window, we can specify -side =>
               'left'.


               Using
our Hello World example as a base, let's look at what happens
when we pack our Button against the different
sides. In Figure 2-5, the only lines we change are
the ->pack part of the Button creation line and
the "Hello World" string in the
$mw->title command to easily show the new
options to pack.

[image: Packing a button against different sides]

Figure 2-5. Packing a button against different sides

The windows in Figure 2-5 have been made a bit
larger to emphasize the difference that using alternative values for
-side makes. Normally, the window will be only as
large as required to show the Button. When you are deciding how to
place widgets in a window, it is always a good idea to see what
happens when you make the window both larger and smaller. Make sure
the behavior you get is what you want.

So far, pack seems pretty simple, but what if you
want to put more than one Button in your application? What happens
when we add more Buttons?

$mw->Button(-text => 'Done1', -command => sub { exit })->pack;
$mw->Button(-text => 'Done2', -command => sub { exit })->pack;
$mw->Button(-text => 'Done3', -command => sub { exit })->pack;
$mw->Button(-text => 'Done4', -command => sub { exit })->pack;
Since the default -side is top,
we would expect all the Buttons to be mushed up against the top of
the window, right? Sort of. The packer allocates space for each
widget, then manipulates the widget inside that space and the space
inside the window.


               Figure 2-6 shows what the window with the four Done
Buttons looks like; the next section explains why.

[image: Four Buttons packed with default settings]

Figure 2-6. Four Buttons packed with default settings


Allocation Rectangles




               When
given an item to pack, the packer first looks to see which side (top,
bottom, right, or left) to use. It then sets aside an invisible
rectangular area across the length of that side for use only by that
widget.


               In Figure 2-7, the solid-line rectangle represents our empty
window (or Frame), and the dotted-line rectangle is the area the
packer sets aside for the first Button. It actually does go all the
way across the width or height of the window, but to make it easier
to see, it's shown indented slightly.

[image: Rectangular areas set aside by the packer when using -side => `top' and -side => `left']

Figure 2-7. Rectangular areas set aside by the packer when using -side => `top' and -side => `left'

The dimensions for the dotted-line box, which we'll call the
allocation rectangle, are calculated based on
the size of the requesting widget. For both the top and bottom sides,
the allocation rectangle is as wide as the window and only as tall as
the widget to be placed in it. For the right and left sides, the
allocation rectangle is as tall as the window but only as wide as
required to fit the widget.

Our examples so far have used Buttons in which the text of the Button
determines its width. If we create a Button with the text
"Done" and one with the text "Done, Finished,
That's it," the second Button is going to be much wider
than the first. When these two Buttons are placed up against either
the right or left side of the window, the second Button has a wider
allocation rectangle than the first. If we place those same two
Buttons against the top and the bottom, the allocation rectangles are
the same height and width, because the window, not the widget,
determines the width.

After the size of the allocation rectangle is determined, the widget
is placed within the allocation rectangle according to other options
passed and/or the default values of those options. We will go over
those options and how they can affect the allocation rectangle later.

Once the first widget has been placed in the window, the amount of
area available for subsequent allocation rectangles is smaller,
because the first allocation rectangle has used some of the space
(see Figure 2-8).

[image: Second allocation rectangle when default side `top' is used]

Figure 2-8. Second allocation rectangle when default side `top' is used


               When more Buttons are placed against
different sides in the same window, the results will vary depending
on the order.

We'll start by placing one Button along the top, one along the
bottom, and then Buttons on the right and left:

$mw->Button(-text => "TOP", -command => sub { exit })
   ->pack(-side => 'top');

$mw->Button(-text => "BOTTOM", -command => sub { exit })
   ->pack(-side => 'bottom');

$mw->Button(-text => "RIGHT", -command => sub { exit })
   ->pack(-side => 'right');

$mw->Button(-text => "LEFT", -command => sub { exit })
  ->pack(-side => 'left');

               Figure 2-9 shows the allocation rectangles for this
window.

[image: Allocation rectangles for four Buttons]

Figure 2-9. Allocation rectangles for four Buttons


               Figure 2-10 shows what the actual window looks like,
both normal size and resized so it's a bit larger.

[image: Four Buttons placed around the sides of the window]

Figure 2-10. Four Buttons placed around the sides of the window


Filling the Allocation Rectangle




               
               Normally,
the widget is left at the default size, which is usually smaller than
the allocation rectangle created for it. If the
-fill option is used, the widget will resize
itself to fill the allocation rectangle according to the value given.
The possible values are:

-fill => 'none' | 'x' | 'y' | 'both'
Using the value 'x' will resize the widget in the
x direction. Likewise, 'y' will cause the widget
to resize in the y direction. Using -fill
               => 'both' is a good way to
see exactly what size and placement was given to the allocation
rectangle, because 'both' resizes the widget in
both x and y directions. Using our four-Button example again,
we'll specify -fill =>
               'both':

$mw->Button(-text => "TOP", -command => sub { exit })
  ->pack(-side => 'top', -fill => 'both');
 
$mw->Button(-text => "BOTTOM", -command => sub { exit })
  ->pack(-side => 'bottom', -fill => 'both');

$mw->Button(-text => "RIGHT", -command => sub { exit })
  ->pack(-side => 'right', -fill => 'both');

$mw->Button(-text => "LEFT", -command => sub { exit })
  ->pack(-side => 'left', -fill => 'both');

               Figure 2-11 shows the resulting window.
[image: Four Buttons packed to each side using -fill => `both']

Figure 2-11. Four Buttons packed to each side using -fill => `both'

If we switch the Button we create first, we get a different result.
The window in Figure 2-12 was created by packing the
widgets in this order: left, right, top, bottom.

[image: Four Buttons packed to each side in a different order using -fill => `both']

Figure 2-12. Four Buttons packed to each side in a different order using -fill => `both'


               Figure 2-13 demonstrates yet another order, which
really shows that the allocation rectangles change size depending on
which is packed first.

[image: Four Buttons packed in order of top, right, bottom, and left]

Figure 2-13. Four Buttons packed in order of top, right, bottom, and left


               
               A

               common use of -fill is
on widgets with Scrollbars: Listbox, Canvas, and Text. Usually the
Scrollbars are along the edge of the window and you want the Listbox
to fill the remaining area. See Chapter 6
and Chapter 7 for more information.


Expanding the Allocation Rectangle




               
               The -expand
option manipulates the allocation rectangle and not the widget inside
it. The value associated with -expand is a Boolean
value.

-expand => 1 | 0
Given a true value, the allocation rectangle will expand into any
leftover space in the window, depending on which side the widget was
packed.

Widgets packed with side 'right' or
'left' will expand in the horizontal direction.
Widgets packed with side 'top' or
'bottom' will expand in the vertical direction. If
more than one widget is packed with -expand turned
on, the extra space in the window is divided evenly among all the
allocation rectangles that want it.

In Figure 2-11 and Figure 2-12, you saw that there was some space left in the
center of the window that wasn't occupied by any widget. If we
change the code and add -expand =>
               1 to the list of pack options for each Button, the
result is the window in Figure 2-14.

[image: Four Buttons using the -expand => 1 and -fill => `both' options]

Figure 2-14. Four Buttons using the -expand => 1 and -fill => `both' options

Note that in Figure 2-14, fill
               => 'both' is left in the
code. If we omit the -fill option, the Buttons
stay their original sizes, but the allocation rectangles (which are
invisible) take over the extra space in the window (see Figure 2-15). 

[image: Four Buttons using -expand => 1 and -fill => `none']

Figure 2-15. Four Buttons using -expand => 1 and -fill => `none'

In Figure 2-15, the Buttons are centered in their
allocation rectangles because of the default value of the
-anchor option, which is
'center'.


Anchoring a Widget in Its Allocation Rectangle




               
               The
-anchor option manipulates the widget inside the
allocation rectangle by anchoring it to the place indicated by the
value passed in. It uses the points of a compass as references.

-anchor =>  'e' | 'w' | 'n' | 's' | 'ne' | 'nw' | 'se' | 'sw' | 'center'

               Figure 2-16 shows those locations in an example
allocation rectangle.

[image: Allocation rectangle with -anchor points labeled]

Figure 2-16. Allocation rectangle with -anchor points labeled

The default for -anchor is
'center', which keeps the widget in the center of
its allocation rectangle. Unless the -expand
option is set to a true value, this won't seem to change much
of anything in the window. As seen in Figure 2-17,
which shows the result of using the -expand
               => 1 option, it is obvious
that the widget sticks to that center position when the window is
resized.

[image: Default behavior of -anchor with -expand set to 1]

Figure 2-17. Default behavior of -anchor with -expand set to 1

If all other defaults are used to pack the widget, Figure 2-18 shows what -anchor =>
               'e' and -anchor => 'w' do.

[image: Examples of -anchor => `e' and -anchor => `w']

Figure 2-18. Examples of -anchor => `e' and -anchor => `w'


               Remember that the allocation rectangle is
created based on which side the widget is packed against, so certain
combinations will appear not to have had any effect. For example:

$mw->Button(-text => "Done", -command => sub { exit })
  ->pack(-side => 'top', -anchor => 'n');

               This code fragment will leave the widget
exactly where it was if the -anchor option had not
been specified, because the allocation rectangle does not change size
at all. If the -expand option is also specified,
when the window is resized, the widget sticks to the north side of
the window. If -anchor =>
               's' is specified, when the window is resized, the
widget sticks to the south side of the window.

The -anchor option is more often used to line up
several widgets in a row. Figure 2-19 and Figure 2-20 show two common examples.

[image: Window with three Buttons all packed with -side => `top', -anchor => `w']

Figure 2-19. Window with three Buttons all packed with -side => `top', -anchor => `w'

[image: Window with three Buttons all packed with -side => `left', -anchor => `n']

Figure 2-20. Window with three Buttons all packed with -side => `left', -anchor => `n'

Sometimes when -side and
-anchor are used together, the results don't
seem to be what you would expect at first glance. Always keep in mind
that invisible allocation rectangle and how it affects what you see
on the screen.
                
            

Widget Order in the Window




               
               
               
               Each window into which widgets are packed
keeps track of those widgets in an ordered list. The order of this
list is determined by the order in which the widgets were packed; the
last item packed is the last item in the list. Using the
-after option, you can change the default order by
specifying which widgetyour new widget should be placed after. On
the opposite end, if you use the -before option,
you can put the new widget before a previously packed widget:

-after => $otherwidget
-before => $otherwidget
As an example, let's create four Buttons
($widget1, $widget2,
$widget3, and $widget4) and
only pack three to begin with. The pack command
for $widget4 might then be:

$widget4->pack(-after => $widget1);

               Figure 2-21 shows two windows: one before
$widget4 is packed and one after
$widget4 is packed.

[image: The window on the right has the Done4 Button packed using -after => $widget1]

Figure 2-21. The window on the right has the Done4 Button packed using -after => $widget1

If we want to put $widget4 in front of
$widget1, we use this command:

$widget4->pack(-before => $widget1);

               Figure 2-22 shows the results. 
[image: Button with Done4 label packed using -before => $done1]

Figure 2-22. Button with Done4 label packed using -before => $done1


Padding the Size of the Widget




               
               
               The final way to force
pack to size the widget is to use the padding
options. The first set of padding options affects the widget itself
by adding to its default size. Different amounts can be added in the
x and y directions, or they can be the same. To specify how much
padding should occur in the x direction, use the
-ipadx option:

-ipadx => amount
Specify padding for the y direction like this:
-ipady => amount
The amount is a
number that is a valid screen distance. We'll discuss the
definition of a valid screen distance in the next section.

Both the -ipadx and -ipady
options change the size of the widget before the allocation rectangle
is calculated. -ipadx adds the amount specified to
both the right and left sides of the widget. The overall width of the
widget is increased by 2 x amount.
-ipady adds to the top and bottom of the widget,
causing the overall height of the widget to increase by 2x
amount. Figure 2-23 shows how the -ipadx and
-ipadyoptions affect a
Button.

[image: The Done1 Button was created with options -ipadx => 10, -ipady => 10]

Figure 2-23. The Done1 Button was created with options -ipadx => 10, -ipady => 10

The other kind of padding is inserted between the edge of the widget
and the edge of the allocation rectangle and is done with the
-padx and -pady options:

-padx => amount
-pady => amount
Using -padx and -pady does not
affect the size of the widget, but it does affect the size of the
allocation rectangle. It acts as a buffer around the widget,
protecting it from touching other widgets. Figure 2-24 shows the effects of using
-padx and -pady.

[image: The Done1 Button was created with options -padx => 10, -pady => 10]

Figure 2-24. The Done1 Button was created with options -padx => 10, -pady => 10

A good way to remember the difference between
-ipadx/y and -padx/y is that
the "i" stands for "inside the widget" or
"internal padding."

Valid screen distances




                  
                  Many times you'll see options that
require values specified in screen units (or what is called a valid
screen distance). The options -ipadx and
-ipady are examples of this type of option. Always
check to see what value the option actually requires.

A screen unit is a number followed by a designation for the unit to
use. If there is no designation, the units are in pixels. Table 2-1 shows all the possibilities.

Table 2-1. Valid screen units
	
                              Designator

                           	
                              Meaning

                           	
                              Examples

                           
	
                              (None)

                           	
                              Pixels (default)

                           	
                              
                                 20, 30,
"20", "40"
                              

                           
	
                              
                                 c
                                 
                              

                           	
                              Centimeters

                           	
                              
                                 '3c', '4c',
"3c"
                              

                           
	
                              
                                 i
                              

                           	
                              Inches

                           	
                              
                                 '2i', "3i"
                              

                           
	
                              
                                 m
                              

                           	
                              Millimeters

                           	
                              
                                 '4m', "4m"
                              

                           
	
                              
                                 p
                              

                           	
                              Printer points (1/72 inch)

                           	
                              
                                 "72p", '40p'
                              

                           




                  
                  
                  
                  To use these designators, it is necessary
to use quotes (either single or double) around the value. Here are
some examples:

$button->pack(-ipadx => 20);      # 20 pixels
$button->pack(-ipadx => '20');    # Also 20 pixels
$button->pack(-ipadx => "1i");    # 1 inch
$button->pack(-ipadx => '1m');    # 1 millimeter
$button->pack(-ipadx => 1);       # 1 pixel
$button->pack(-ipadx => "20p");   # 20 printer points
Remember that a p designator does not stand for
pixels, but printer points.



Displaying in a Parent Other Than Your Own




               By default,
when a widget is packed, it is packed inside the region that created
it. Sometimes it is necessary to display a widget inside a different
region. Use the -in option to do so:

-in => $otherwindow
It puts the new widget at the end of the packing order for
$otherwindow and displays it accordingly. All
other options specified in the pack call still
apply.


Methods Associated with pack




               There are a few methods that
are used in conjunction with the pack geometry
manager. They allow the programmer to get information about either
the widget that has been packed or the parent widget in which other
widgets are packed.

Unpacking a widget




                  
                  
                  To unpack a widget from a window
or Frame, use the packForget method:

$widget->packForget( );

                  packForget makes it look like the widget
disappears. The widget is not destroyed, but it is no longer managed
by pack. The widget is removed from the packing
order, so if it's repacked later, it appears at the end of the
packing order.


Retrieving pack information



To return a list containing all the pack
configuration information about a widget, use
packInfo:

@list = $widget->packInfo(  );

                  
                  The
format of the list is in option/value pairs. The first pair in the
list is -in and the current window that contains
$widget (usually also the parent). Here's an
example of the information returned from packInfo:

-in MainWindow=HASH(0x818dcf4) -anchor n -expand 0 -fill none -ipadx 0 -ipady 0 -padx 
10 -pady 10 -side left
From this we can tell that we packed our $widget
into the MainWindow rather than into a Frame. Since the list has a
"paired" quality to it, we could easily store the result
from packInfo in a hash and reference the
different option values by using a key to the hash:

%packinfo = $widget->packInfo;
print "Side used: ", $packinfo{-side}, "\n";

Disabling and enabling automatic resizing



Unless you've set a preferred window size via the
geometry method explicitly, when you put a widget
inside a window, the window (or Frame) will resize itself to
accommodate the widget. If you are placing widgets inside your window
dynamically while the program is running, the window will appear to
bounce from size to size. You can turn this behavior off by using
packPropagateon the Frame or
Toplevel widget:

$widget->packPropagate(0);
If set to 0 or 'off',
packPropagate changes the behavior of the widget
so it doesn't resize to accommodate items packed inside of it.
When a false value is sent to packPropagate before
widgets are placed inside it, this automatic resizing doesn't
happen, so you can't see any of the widgets placed inside the
parent until it is manually resized. If you call
packPropagate after the widgets have been placed
inside it, the widget will ignore any size changes from its child
widgets.


Listing widgets




                  
                  
                  
                  You
can determine the widgets your Frame or Toplevel holds with the
packSlaves method:

@list = $parentwidget->packSlaves( );

                  packSlaves returns an ordered list of all the
widgets that were packed into $parentwidget. An
empty string (or empty list) is returned if no widgets were packed
into $parentwidget.

The list returned from packSlaves looks like this:
Tk::Button=HASH(0x81b2970) Tk::Button=HASH(0x8116ccc) Tk::Button=HASH(0x81bcdd4)

                  Each
item is a reference to a packed widget and can be used to configure
it. For example, you can increase the size of each widget by 20 in
both the x and y directions by looping through it and
"packing" it with new information. Using our good window
example in Figure 2-4, we can add a Button that
will contain a subroutine that uses packSlaves:

$mw->Button(-text => "Enlarge",
            -command => \&repack_kids)->pack(-side => 'bottom',
                                             -anchor => 'center');
sub repack_kids {
  my @kids = $mw->packSlaves;
  foreach (@kids) {
    $_->pack(-ipadx => 20, -ipady => 20);
  }
}

                  Figure 2-25 shows the resulting window.
[image: Window before pressing Enlarge Button]

Figure 2-25. Window before pressing Enlarge Button 

Let's look at what happens when we press the Enlarge Button. As
shown in Figure 2-26, all the widgets are now
repacked with additional parameters of -ipadx
                  => 20,
                  -ipady =>
                  20. These new options are in addition to any other
parameters with which the widgets were packed previously. If an
option is repeated, the last one specified overrides the previous
ones.

[image: Window after pressing Enlarge Button]

Figure 2-26. Window after pressing Enlarge Button

The window is suddenly huge! Subsequent presses of the Enlarge Button
will do nothing more to the window, because each of the widgets
already has an -ipadxand
-ipady of 20. If we wanted to always add 20 to the
values of -ipadx and -ipady, we
would have to request the current values and add 20 to them:

sub repack_kids {
  my @kids = $mw->packSlaves;
  foreach (@kids) {
    %packinfo = $_->packInfo(  );
    $_->pack(-ipadx => 20 + $packinfo{"-ipadx"}, 
             -ipady => 20 + $packinfo{"-ipady"});
  }
}
We use packInfo to get the current configuration
and add 20 to that value.



Demo Programs for pack




               
               Here are three little programs that let
you experiment with the packing options of an unspecified number of
widgets.

As you can see in Figure 2-27, there is both a
console window and an "output" window that show what is
happening to widgets as you change their options. The complete code
follows shortly.

[image: The output and console windows for our Play With pack example]

Figure 2-27. The output and console windows for our Play With pack example

There are a lot of widgets and methods in the code that we
haven't covered yet. For now, it would be useful to download
the code, run it, and play around with the options.

use Tk;
require Tk::BrowseEntry;

if ($#ARGV >= 0) { $numWidgets = $ARGV[0]; }
else { $numWidgets = 4; }

$mw = MainWindow->new(-title => "Play w/pack");
$f = $mw->Frame(-borderwidth => 2, -relief => 'groove')
	->pack(-side => 'top', -fill => 'x');
my (@packdirs) = (  );

foreach (0..$numWidgets)
{
	$packdirs[$_] = 'top';
	my $be = $f->BrowseEntry(-label => "Widget $_:", 
		-choices => ["right", "left", "top", "bottom"], 
		-variable => \$packdirs[$_], -browsecmd => \&repack)
		->pack(-ipady => 5);
}

$f->Button(-text => "Repack", -command => \&repack )
	->pack(-anchor => 'center');

# use a separate window so we can see what the output 
# looks like without clutter.
$top = $mw->Toplevel(-title => "output window");
my $c;
foreach (@packdirs)
{
	my $b = $top->Button(-text => $c++ . ": $_", 
		-font => "Courier 20 bold")
		->pack(-side => $_, -fill => 'both', -expand => 1);
}

MainLoop;

sub repack
{
	@w = $top->packSlaves;
	foreach (@w) { $_->packForget; }
	my $e = 0;
	foreach (@w)
	{
		$_->configure(-text => "$e: $packdirs[$e]");
		$_->pack(-side => $packdirs[$e++], -fill => 'both', -expand => 1)	;
	}
}
A more complicated version of our pack demo lets
us alter not just the side the widget is packed against but the more
commonly used pack options:

use Tk;
require Tk::BrowseEntry;

if ($#ARGV >= 0) { 	$numWidgets = $ARGV[0]; }
else { 	$numWidgets = 1; }

$mw = MainWindow->new(-title => "Play w/pack");

$f = $mw->Frame(-borderwidth => 2, -relief => 'groove')
	->pack(-side => 'top', -fill => 'x');

# Initialize the variables
my (@packdirs) = (  );
my (@anchordirs) = (  );
my (@fill) = (  );
my (@expand) = (  );

$i = 0;
$top = $mw->Toplevel(-title => "output window");
my $addbutton = $f->Button(-text => "Add Widget", 
	-command => \&addwidget )->pack(-anchor => 'center');

foreach (0..$numWidgets) {
ch04	
	&addwidget($_);	
}
MainLoop;

sub repack {
	print "Repacking...";
	@w = $top->packSlaves;
	foreach (@w) { $_->packForget; }
	my $e = 0;
	foreach (@w) 	{
		$_->configure(-text => "$e: $packdirs[$e]");
		print "Expand is : " . $expand[$e]. "\n";
		$_->pack(-side => $packdirs[$e], 
						 -fill => $fill[$e], 
						 -expand => $expand[$e], 
						 -anchor => $anchordirs[$e]);
		$e++;
	}
}

sub addwidget {
	my ($count) = @_;
	print "COUNT $count\n";
	if (! defined $count) 	{
		$numWidgets ++;
		$count = $numWidgets ;
	}
	
	$packdirs[$count] = 'top';
	$anchordirs[$count] = 'center';
	$fill[$count] = 'none';
	$expand[$count] = 0;
	
	my $f1 = $f->Frame->pack(-side => 'top', -expand => 1, 
		-fill =>'y', -before => $addbutton);
	my $be = $f1->BrowseEntry(-label => "Widget $count:", 
		-choices => ["right", "left", "top", "bottom"], 
		-variable => \$packdirs[$count], -browsecmd => \&repack)
		->pack(-ipady => 5, -side => 'left');

	$f1->BrowseEntry(-label => "-anchor", 
		-choices => [qw/center n s e w ne se nw sw/], 
		-variable => \$anchordirs[$count], -browsecmd => \&repack)
		->pack(-ipady => 5, -side => 'left');

	$f1->BrowseEntry(-label => "-fill", -choices => [qw/none x y both/], 
		-variable => \$fill[$count], -browsecmd => \&repack)
		->pack(-ipady => 5, -side => 'left');

	$f1->Checkbutton(-text => "-expand", -onvalue => 1, -offvalue => 0,
		-variable => \$expand[$count], -command => \&repack)
		->pack(-ipady => 5, -side => 'left');

	$top->Button(-text => $count . ": $packdirs[$count]",
		-font => "Courier 20 bold")->pack(-side => $packdirs[$count], 
                -anchor => $anchordirs[$court]
		-fill => $fill[$count], -expand => $expand[$count]);
}

               Figure 2-28 shows what the new console looks like
(the output window looks the same until you vary the options).

[image: The console for the more complicated version of Play With pack]

Figure 2-28. The console for the more complicated version of Play With pack

This sample switches packed widgets between a visible and invisible
state. As the widgets are created, save their references as the array
@w. The scalar $packed is
associated with the Checkbutton. Whenever we click the Checkbutton,
the value of $packed toggles between 1 and 0. The
first click of the Checkbutton sets $packed to 0
and invokes the -command callback. For more
information on callbacks, see Chapter 15.

The callback then removes the widgets from the display, using
packInfo to fetch their pack
attributes, which are stored in an instance variable. Refer to Chapter 14 for details on instance variables. When
restoring widgets, the callback uses the saved
pack information to exactly replicate the initial
packing configuration.
               
                
            
my $f = $mw->Frame->pack;
my $packed = 1;

push  my @w, $f->Label(-text => "l1")->pack(-side => "bottom"),
  $f->Button(-text => "l2")->pack(-side => "right"),
  $f->Label(-text => "l3")->pack(-side => "top"),
  $f->Button(-text => "l4")->pack(-side => "left"),
;
$mw->Checkbutton(-text => "Pack/Unpack",
		 -variable => \$packed,
		 -command => sub {
		     if ($packed) {
			 foreach (@w) {
			     $_->pack(@{ $_->{PackInfo} });
			 }
		     } else {
			 foreach (@w) {
			     $_->{PackInfo} = [ $_->packInfo ];
			     $_->packForget;
			 }
		     }
		 })->pack;



The grid Geometry Manager




            The
grid geometry manager divides the window into a
grid composed of columns and rows starting at (0, 0) in the
upper-left corner. Figure 2-29 shows a sample grid.

[image: A window divided into grids]

Figure 2-29. A window divided into grids


            Rather than using the sides of a
window as reference points, grid divides the
screen into columns and rows. It looks a lot like a spreadsheet,
doesn't it? Each widget is assigned a grid cell using the
options available to grid.


            The
grid method takes a list of widgets instead of
operating on only one widget at a time.[1] Here is the generic usage:

$widget1->grid( [ $widget2, ... , ] [ option => value, ... ] );
A specific example is:
$widget1->grid($widget2, $widget3);
Instead of using three separate calls, you can use one
grid call to display all three widgets. You can
also invoke grid on each widget independently,
just as you can pack. Each call to
grid will create another row in the window. So in
our example, $widget1,
$widget2, and$widget3 will be placed in the first
row. Another call to grid creates a second row.
This is what happens when you do not specify any additional options
to the grid call.

The previous example can be rewritten like this:
Tk::grid($widget1, $widget2, $widget3);

            But beware, this is not necessarily
equivalent to the previous statement, due to
inheritance, an object-oriented concept. For
more information, please refer to Chapter 14.
Essentially, using Tk::grid is the same as calling
a subroutine directly, whereas the method call searches the
widget's class hierarchy for a subroutine
grid. It's certainly possible that
$widget1 has its own special
grid method, which we would rudely bypass. Is this
a likely possibility? No. Just be aware when you make a procedural
versus a method call.


            
            For
greater control, you can specify explicit -row and
-column options for each widget in the window.
We'll cover these options later.

When additional options are not specified, the following assumptions
are made:

	The first widget in the row (e.g., $widget1 in the
preceding example) invokes the grid command.


	All remaining widgets for that row will be specified as arguments to
the grid command.


	Each additional call to gridwill add another row to the display.


	Special characters can be used to change the
-columnspan and -rowspan of the
widget without using -columnspan or
-rowspan explicitly.




A few examples will help demonstrate. Each call to
grid creates another row, so in the following
example we have two rows:

# Create two rows, each with four widgets
$widget1->grid($widget2, $widget3, $widget4);
$widget5->grid($widget6, $widget7, $widget8);
In this example, we have created four rows and there is only one
widget in each row:

# Create four rows, each with one widget
$widget1->grid(  );
$widget2->grid(  );
$widget3->grid(  );
$widget4->grid(  );
We can also create widgets as we go:
$mw->Button(-text => 'Button1', -command => \&call1)->grid(
            $mw->Button(-text => 'Button2', -command => \&call2),
            $mw->Button(-text => 'Button3', -command => \&call3),
            $mw->Button(-text => 'Button4', -command => \&call4));
Pay careful attention, because the second, third, and fourth calls to
Button are inside the call to
grid. All four of the Buttons will be placed in
the first row. If we execute the same command again, the new widgets
are placed in the next row.

Special Characters




               There are several special characters
that can be used to alter the way the widgets are gridded in the
window. Each special character serves as a type of placeholder that
indicates what to do with that position in the grid:

	
                     "-" (a minus sign)
	
                        
                        Tells grid that the widget specified just
before this one in the list should span this column as well. To span
more than one column, place a "-" in each widget
position to span. A "-" cannot follow a
"^" or an "x".


	
                     "x"
                  
	
                        Effectively leaves a blank space
where a widget would otherwise be placed.


	
                     "^"
                  
	
                        
                        A widget in row x
will span row x and x + 1
when this character is placed in the grid command
for row x + 1 in that row/column position. The
number of "^" characters must match the number of
columns the widget spans in row x. Similar to
"-", but goes down, not across.




The following sections include some examples that illustrate what the
special characters do.

Spanning columns




                  
                  The following bit of code creates three
rows of Buttons. The first two rows are normal and, in the third, the
second Button spans three columns. Each "-"
character adds one to the number of columns the Button uses, and the
default is one. So the original column and two hyphens
("-","-") indicate that there
are three columns to span. The -sticky option is
necessary for the widgets to stick to the sides of the cells they
span. If the -sticky option were left out, the
Button would be centered across the three cells it spans.

$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }));

$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ($mw->Button(-text => "Button6", -command => sub { exit }),
   $mw->Button(-text => "Button7", -command => sub { exit }),
   $mw->Button(-text => "Button8", -command => sub { exit }));


$mw->Button(-text => "Button9", -command => sub { exit })->grid
  ($mw->Button(-text => "Button10", -command => sub { exit }),
   "-", "-",  -sticky => "nsew");

                  Figure 2-30 shows the resulting window.
[image: Example of column spanning using the "-" character]

Figure 2-30. Example of column spanning using the "-" character


Empty cells




                  
                  
                  The "x"
character translates to "skip this space" and leaves a
hole in the grid. We removed the line that created Button6 and
replaced it with an "x" in the following code. The
cell is still there, it just doesn't contain a
widget.
               
$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }));
 
$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ("x",
   $mw->Button(-text => "Button7", -command => sub { exit }),
   $mw->Button(-text => "Button8", -command => sub { exit }));
 
                  Figure 2-31 shows the resulting window.
[image: Leaving an empty cell between widgets]

Figure 2-31. Leaving an empty cell between widgets



grid Options




               The rest of the options are similar to
those used with pack:

	
                     "-"
                  
	A special character used in the grid widget list.
Increases columnspan of the prior widget in the
widget list.


	
                     "x"
                  
	A special character used in the grid widget list.
Leaves a blank space in the grid.


	
                     "^"
                  
	
                        
                        A special character used in the
grid widget list. Increases
rowspan of the widget in the grid directly above
it.


	
                     -column => 
                     n
                  
	
                        Sets the column to place the widget in
(
                        n 
                        >=
0). 


	
                     -row => 
                     m
                  
	
                        Sets the row to place the widget in
(
                        m 
                        >=
0). 


	
                     -columnspan => 
                     n
                  
	
                        
                        Sets
the number of columns for the widget to span beginning with
-column.


	
                     -rowspan => 
                     m
                  
	
                        Sets the
number of rows for the widget to span beginning with
-row.


	
                     -sticky => 
                     string
                  
	
                        Sticks the widget to
string sides. String contains characters
n, s, e, or
w.


	
                     -in => $otherwindow
                  
	
                        Indicates the widget is gridded inside
$otherwindow instead the parent of
$widget.


	
                     -ipadx => 
                     amount
                  
	
                        $widget
                        
becomes larger in x direction by amount.


	
                     -ipady => 
                     amount
                  
	
                        $widget becomes larger in y direction by
amount.


	
                     -padx => 
                     amount
                  
	Places buffer space equal to amount to the
left and right of the widget.


	
                     -pady => 
                     amount
                  
	
                        
                        Places buffer space equal to
amount on the top and bottom of the
widget.





Specifying Rows and Columns Explicitly




               
               Rather than
letting grid make assumptions, it is sometimes
necessary to explicitly state the row and column in which the widget
should be placed. This is done by using the -row
and -column options. Each option takes a
nonnegative integer as an argument:

-column => n, -row => m
When you use -row and -column,
it is not necessary to build or grid the widgets
in any sort of logical order (except for your own sanity when you are
debugging). You could place your first widget in column 10 and row 5
if you like. All the other cells with lower row and column values
will remain empty.


Spanning Rows and Columns Explicitly




               
               
               It
is also possible to indicate explicitly that a widget (or widgets)
should span some columns or rows. The option to span columns is
-columnspan. For spanning rows, the option is
-rowspan
               . Both options take
an integer that is 1 or greater. The value indicates how many rows or
columns should be spanned, including the row or column in which the
widget is placed.

For this example, we use the easy way to place widgets in columns and
rows by not explicitly specifying the -row and
-column options. Note that the second
grid command applies to two Button widgets, so the
single -columnspan option applies to
both Buttons created there.

$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }),
   -sticky => "nsew");

# Button5 will span Columns 0-1 and Button6 will span 2-3
$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ($mw->Button(-text => "Button6", -command => sub { exit }),
   -sticky => "nsew", -columnspan => 2);
The resulting window is shown in Figure 2-32.
[image: Nonexplicit -columnspan example]

Figure 2-32. Nonexplicit -columnspan example

This window could also have been created using the
"-" special character to indicate column spanning,
like this:

$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }),
   -sticky => "nsew");

# Button5 will span Columns 0-1 and Button6 will span 2-3
$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ("-", $mw->Button(-text => "Button6", -command => sub { exit }), "-"
   -sticky => "nsew");
This example illustrates how to explicitly use the
-row and -column options in
addition to the -rowspan option:

$mw->Button(-text => "Button1", -command => sub { exit })->
  grid(-row => 0, -column => 0, -rowspan => 2, -sticky => 'nsew');
$mw->Button(-text => "Button2", -command => sub { exit })->
  grid(-row => 0, -column => 1);
$mw->Button(-text => "Button3", -command => sub { exit })->
  grid(-row => 0, -column => 2);
$mw->Button(-text => "Button4", -command => sub { exit })->
  grid(-row => 0, -column => 3);

$mw->Button(-text => "Button5", -command => sub { exit })->
  grid(-row => 1, -column => 1);
$mw->Button(-text => "Button6", -command => sub { exit })->
  grid(-row => 1, -column => 2);
$mw->Button(-text => "Button7", -command => sub { exit })->
  grid(-row => 1, -column => 3);
See Figure 2-33 for the resulting window.
[image: Explicit -rowspan example]

Figure 2-33. Explicit -rowspan example


Forcing a Widget to Fill a Cell




               
               When you use the
pack command, it is necessary to indicate both
-fill and -expand options to
get the widget to resize inside its allocation rectangle. The
gridcommand doesn't
have an allocation rectangle to fill, but it does have the cell
within the grid. Using the -sticky option with
grid is similar to using -fill
and -expand with pack.


               
               
               
               The value associated with
-sticky is a string containing the compass points
to which the widget should "stick." If the widget should
always "stick" to the top of the cell, you would use
-sticky =>
               "n". To force the widget to fill the cell
completely, use -sticky =>
               "nsew". To make the widget as tall as the cell but
only as wide as it needs to be, use -sticky
               => "ns". The string value
can contain commas and whitespace, but they will be ignored. These
two statements are equivalent:

-sticky => "nsew"
-sticky => "n, s, e, w"  # Same thing

               
               If you use
-sticky with your widgets and then resize the
window, you'll notice that the widgets don't resize as
you would expect. This is because resizing of the cells and the
widgets in them is taken care of with the
gridColumnconfigure and
gridRowconfigure methods, which are discussed
later in this chapter.


Padding the Widget




               grid
               
also accepts these four options: -ipadx,
-ipady, -padx, and
-pady. They work exactly the same as they do in
pack, but instead of affecting the size of the
allocation rectangle, they affect the size of the cell in which the
widget is placed.


               
               In
this example, the -ipady and
-ipadx options are applied to the top row of
Buttons and not the bottom row:

$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }),
   -sticky => "nsew", -ipadx => 10, -ipady => 10);

$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ($mw->Button(-text => "Button6", -command => sub { exit }),
   $mw->Button(-text => "Button7", -command => sub { exit }),
   $mw->Button(-text => "Button8", -command => sub { exit }),
   -sticky => "nsew");
Notice in Figure 2-34 how Buttons 5 through 8 are
also wider than they really need to be. This is because we used the
-sticky =>
               "nsew" option.

[image: grid -ipadx and -ipady example]

Figure 2-34. grid -ipadx and -ipady example


               
               In
the following example, the -pady and
-padx options are applied to the top row of
Buttons and not the bottom row. Figure 2-35 shows
the results.

$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }),
   -sticky => "nsew", -padx => 10, -pady => 10);

$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ($mw->Button(-text => "Button6", -command => sub { exit }),
   $mw->Button(-text => "Button7", -command => sub { exit }),
   $mw->Button(-text => "Button8", -command => sub { exit }),
   -sticky => "nsew");
[image: grid -padx and -pady example]

Figure 2-35. grid -padx and -pady example


Specifying a Different Parent




               
               
               The
-inoption works the same way
in grid as it does in pack.
$widget will be placed in
$otherwindow and not in the default parent of
$widget.

Here is the usage:
-in => $otherwindow

Configuring Columns and Rows




               
               
               As with any of the geometry managers,
grid has a few methods associated with it. Each
method is invoked via a widget that has been placed on the screen by
using grid. Sometimes it is necessary to change
the options of the group of cells that makes up your grid.


               
               You
can control resizing and the minimum size of a cell with the
gridColumnconfigure and
gridRowconfigure methods. Each takes a column or a
row number as its first argument and then takes some optional
arguments that will change the configuration of that column or row.


               
               
               
               
               Both
gridColumnconfigure and
gridRowconfigure work similarly to the
configure method used with widgets; however, the
options you can specify with gridColumnconfigure
and gridRowconfigure cannot be used with the
grid command. The options you can use with
gridColumnconfigure and
gridRowconfigure are -weight,
-minsize, and -pad.

If you send only a row or column number, a list of key/value pairs is
returned with the current options and their values for that method:

@column_configs = $mw->gridColumnconfigure(0);
@row_configs = $mw->gridRowconfigure(0);
Depending on your sensibilities, you may want to store the results in
a hash:

%column_configs = $mw->gridColumnconfigure(0);
%row_configs = $mw->gridRowconfigure(0);
In this example, we are getting the options and their values for the
first column and the first row. The results of using the default
values would look like this:

-minsize 0 -pad 0 -weight 0
-minsize 0 -pad 0 -weight 0
You can get the value of only one of the options by sending that
option as the second argument:

print $mw->gridColumnconfigure(0, -weight), "\n";
print $mw->gridRowconfigure(0, -weight), "\n";
The results would be:
0
0
To change the value of the option, use the option followed
immediately by the value you want associated with it. For example:

$mw->gridColumnconfigure(0, -weight => 1);
$mw->gridRowconfigure(0, -weight => 1);
You can also specify multiple options in one call:
$mw->gridColumnconfigure(0, -weight => 1, -pad => 10);
$mw->gridRowconfigure(0, -weight => 1, -pad => 10);
Now that we know how to call gridColumnconfigure
and gridRowconfigure, we need to know what the
three different options do.

Weight




                  
                  The -weight option
sets the amount of space allocated to the column or row when the
window is divided into cells. Remember to use
-sticky =>
                  "nsew" in your grid command if
you want the widget to resize when the cell does. The default
-weight is 0, which causes the column width or row
height to be dictated by the largest widget in the column. Each
-weight value has a relationship to the other
-weights in the rows or columns.

If a column or row has a -weight of 2, it is twice
as big as a column or row that has a -weight of 1.
Columns or rows of -weight 0 don't get
resized at all. If you want all your widgets to resize in proportion
to the size of the window, add this to your code before you call
MainLoop:

($columns, $rows) = $mw->gridSize(  );
for ($i = 0; $i < $columns; $i++) {
  $mw->gridColumnconfigure($i, -weight => 1);
}
for ($i = 0; $i < $rows; $i++) {
  $mw->gridRowconfigure($i, -weight => 1);
}
This code will assign the -weight of 1 to every
single row and column in the grid, no matter what size the grid is.
Of course, this example works only if you want to assign the same
size to each row and each column, but you get the idea.

Here is an example of how the -weight option works
(Figure 2-36 shows the result):

$mw->Button(-text => "Button1", -command => sub { exit })->grid
  ($mw->Button(-text => "Button2", -command => sub { exit }),
   $mw->Button(-text => "Button3", -command => sub { exit }),
   $mw->Button(-text => "Button4", -command => sub { exit }),
   -sticky => "nsew");

$mw->Button(-text => "Button5", -command => sub { exit })->grid
  ("x",
   $mw->Button(-text => "Button7", -command => sub { exit }),
   $mw->Button(-text => "Button8", -command => sub { exit }),
   -sticky => "nsew");

$mw->gridColumnconfigure(1, -weight => 1);
$mw->gridRowconfigure(1, -weight => 1);
By giving row 1 and column 1 weights of 1 (whereas all other rows and
columns have 0 weights), they take over any extra available space
when the size of the window is increased. Notice that columns 0, 2,
and 3 are only as wide as is necessary to draw the Buttons and their
text, but column 1 has filled in the extra space. The same effect
happens for row 0 with a weight of 0 and row 1 with a new weight of
1. (The window has been resized larger to demonstrate the effects of
-weight.)

[image: gridRowconfigure and gridColumnconfigure example]

Figure 2-36. gridRowconfigure and gridColumnconfigure example


Minimum cell size




                  
                  The option
-minsize sets the smallest width for the column or
the smallest height for each row. The -minsize
option takes a valid screen distance as a value. In this example, the
minimum size of the cells in row 0 and column 0 is set to 10 pixels:

$mw->gridColumnconfigure(0, -minsize => 10);
$mw->gridRowconfigure(0, -minsize => 10);
If the column or row was normally less than 10 pixels wide, it would
be forced to be at least that large.


Padding



You can add padding around the widget and to the widget by using the
-padx/y and -ipadx/y options.
You can also add a similar type of padding by using the
-pad option with the
gridColumnconfigure and
gridRowconfigure methods. This padding is added
around the widget, not to the widget itself. When you call
gridColumnconfigure, the -pad
option will add padding to the left and right of the widget. Calling
gridRowconfigure with -pad will
add padding to the top and bottom of the widget. Here are two
examples:

$mw->gridColumnconfigure(0, -pad => 10);
$mw->gridRowconfigure(0, -pad => 10);

Bounding box



To find out how large a cell is, you can use the
gridBbox method:

($xoffset, $yoffset, $width, $height) = $master->gridBbox(0, 2);
This example gets the bounding box for column 0 and row 2. All the
values returned are in pixels. The bounding box will change as you
resize the window. The four values returned represent the x offset,
the y offset, the cell width, and the cell height (offsets are
relative to the window or Frame where the widget is gridded). The
bounding box dimensions include any and all padding specified by the
-padx, -pady,
-ipadx, and -ipady
options.
                  
                  
               


Removing a Widget




               
               Like
packForget, gridForget removes
widgets from view on the screen. This may or may not cause the window
to resize itself; it depends on the size of
$widget and where it was on the window. Here are
some examples:

$mw->gridForget(  );                # Nothing happens
$widget->gridForget(  );            # $widget goes away
$widget->gridForget($widget1);    # $widget and $widget1 go away
$widget->gridForget($w1, $w3);    # $widget, $w1, $w3 go away
The widgets are undrawn from the screen, but the cells remain
logically filled.


Getting Information



ThegridInfo method returns
information about the $widget in a list format.
Just as with packInfo, the first two elements
indicate where the widget was placed:

@list = $widget->gridInfo(  );  # Easier to print
%gridInfo = $widget->gridInfo(  );
Here are some sample results from gridInfo:
-in Tk::Frame=HASH(0x81abc44) -column 0 -row 0 -columnspan 1 -rowspan 2 -ipadx 0 
-ipady 0 -padx 0 -pady 0  -sticky nesw

Widget Location



The gridLocation method returns the column and row
of the widget nearest the given (x, y) coordinates, relative to the
master:

($column, $row) = $master->gridLocation($x, $y);
Both $x and $y are in screen
units relative to the master window (in our examples,
$mw). For locations above or to the left of the
grid, -1 is returned.

When given the arguments (0, 0), our application returns this: 
0 0
This indicates that the cell is at column 0 and row 0.
Propagation



There is a gridPropagate method that is similar to
packPropagate:

$master->gridPropagate( 0 );
When given a false value, gridPropagate turns off
geometry propagation, meaning size information is not sent upward to
the parent of $master. By default, propagation is
turned on. If gridPropagate is not given an
argument, the current value is returned.



How Many Columns and Rows?



To find out how large the grid has become after placing numerous
widgets in it, you can use gridSize on the
container widget to get back the number of columns and the number of
rows:

($columns, $rows) = $master->gridSize(  );
The list returned contains the number of columns followed by the
number of rows. In many of the earlier examples, we had a grid size
that was four columns by two rows.

($c, $r) = $f->gridSize(  );    #$c = 4, $r = 2
It is not necessary for a widget to be placed in a column/row for it
to be considered a valid column/row. If you place a widget in column
4 and row 5 by using -row=>5, -column=>4 and
the only other widget is in row 0 and column 0, then
gridSize will return 5 and 6.


gridSlaves



There are two ways to find out which widgets have been put in a
window or Frame: use gridSlaves without any
arguments to get the full list, or specify a row and column. Here are
examples of both:

@slaves = $mw->gridSlaves(  );
print "@slaves\n";
The preceding code might have printed this:
Tk::Button=HASH(0x81b6fb8) Tk::Button=HASH(0x81ba454) Tk::Button=HASH(0x81ba4cc) Tk::
Button=HASH(0x81ba538) Tk::Button=HASH(0x81b6fa0) Tk::Button=HASH(0x81ba5e0) Tk::
Button=HASH(0x81ba6dc) Tk::Button=HASH(0x81ba748)
We could have specified the widget in column 0, row 0:
$widget = $mw->gridSlaves( -row => 0, -column => 0 );
print "$widget\n";
# Might print this: Tk::Button=HASH(0x81b6fb8)

               
               If you specify only the
-row option, you'll get a list containing
only the widgets in that row. The same goes for specifying only
-column; your list will contain only the widgets
in that column. 
            



[1] Several
people have mentioned that pack can also take a
list of widgets. We don't cover this because it is not how
pack is normally used.
               



The place Geometry Manager




            The
place geometry manager is different than
grid or pack. Rather than
referencing against a cell location or a window's side, most of
the time you'll be using a relative form of x and y
coordinates. You can also use place to overlap
portions of widgets, which isn't allowed in either
grid or pack.

Invoking place is similar to calling the other
geometry managers:

$widget->place( [ option => value, . . . ] );
The options specified when you call place affect
how the widgets are put on the screen.

place Options



The following options can be used with place:
	
                     -anchor => 'n' | 'ne' | 'e' | 'se' | 's' | 'sw' | 'w' | '
                     nw
                     ' | 'center'
                  
	Sets the position in the widget that will be placed at the specified
coordinates.


	
                     -bordermode => '
                     inside
                     ' | 'outside' | 'ignore'
                  
	
                        Determines whether or not the border
portion of the widget is included in the coordinate system.


	
                     -height => 
                     amount
                  
	
                        Sets the absolute height of the widget.


	
                     -in => $window
                  
	
                        Indicates that the child widget will be
packed inside $window instead of in the parent
that created it. Any relative coordinates or sizes will still refer
to the parent.


	
                     -relheight => 
                     ratio
                  
	
                        Indicates
that the height of the widget relates to the parent widget's
height by ratio.


	
                     -relwidth => 
                     ratio
                  
	
                        Indicates
that the width of the widget relates to the parent widget's
width by ratio.


	
                     -relx => 
                     xratio
                  
	
                        Indicates that
the widget will be placed relative to its parent by
xratio.


	
                     -rely => 
                     yratio
                  
	Indicates that the widget will be placed relative to its parent by
yratio.


	
                     -width => 
                     amount
                  
	Indicates that the width of the widget will be
amount.


	
                     -x => 
                     x
                  
	Indicates that the widget will be placed at
x. x is any
valid screen distance.


	
                     -y => 
                     y
                  
	Indicates that the widget will be placed at
y. y is any
valid screen distance.





Absolute Coordinates




               
               
               The parent window (or Frame) has a
standard coordinate system where (0, 0) is in the upper-left corner.
The x values increase to the right, and the y values increase as you
go down. See Figure 2-37.

[image: Coordinate system of parent window when absolute coordinates are used]

Figure 2-37. Coordinate system of parent window when absolute coordinates are used


               
               To use absolute coordinates to specify
where to place the widget, we would use options
-xand -y:

-x => x, -y => y

               Valid values
for x and y are valid
screen distances (e.g., 5, which is in pixels). The widget will have
its anchor position (controlled by -anchor) placed
at the x and y coordinates. The default anchor is
"nw", the upper-left corner of the window.

Another major difference between place and the
other geometry managers is that at least two arguments are required
when place is invoked. There are no default values
for the -xand
-y options. You will get an
error if you try to invoke place with no arguments
(for example, $widget->place( )).

The simplest example of using -x and
-y is to place a widget at (0, 0):

$mw->Button(-text => "Exit",
            -command => sub { exit })->place(-x => 0, -y => 0);
As you would expect, the widget ends up in the upper-left corner of
the window as shown in Figure 2-38. No matter what
size the window, our widget will remain positioned at (0, 0). Even
when the window is resized to be as small as possible, the widget
will not move.

[image: Button placed using -x => 0, -y => 0]

Figure 2-38. Button placed using -x => 0, -y => 0

Here is an example of using-x and -y to create
some overlapping widgets:

$mw->Button(-text => "Exit", 
            -command => sub { exit })->place(-x => 10, -y => 10);
$mw->Button(-text => "Exit", 
            -command => sub { exit })->place(-x => 20, -y => 20);

               Figure 2-39 shows the resulting window.
[image: Overlapping Buttons using place]

Figure 2-39. Overlapping Buttons using place


Relative Coordinates




               
               There is an additional coordinate
system defined in place for the parent widget that
allows relative placement within it. This coordinate system is shown
in Figure 2-40.

[image: The relative coordinate system]

Figure 2-40. The relative coordinate system

The upper-left corner has the coordinates (0.0, 0.0). The lower-right
corner's coordinates are (1.0, 1.0). The middle of the window
would be (0.5, 0.5). The coordinates are specified in floating-point
form to allow place to handle any size window.
This allows the widget to remain at that position (in the center, for
instance) no matter how the window is resized.

It is valid to specify coordinates both smaller than 0.0 and larger
than 1.0; however, your widget might not be completely visible in the
window when you use out-of-range coordinates.

This code snippet produces the Button shown in Figure 2-41: 

$b = $mw->Button(-text => "Exit", -command => sub { exit });
$b->place(-relx => 0.5, -rely => 0.5);
[image: Using place with -relx => 0.5, -rely => 0.5]

Figure 2-41. Using place with -relx => 0.5, -rely => 0.5

Although the Button in Figure 2-41 is placed in the
middle of the screen, it looks off-center because the upper-left
corner of the widget was placed in the middle of the window instead
of the center. You can change this with the
-anchor option, which we will discuss shortly. If
we resize this window, the Button still stays in the middle of the
window (see Figure 2-42).

[image: -relx => 0.5, -rely => 0.5 window resized to be larger]

Figure 2-42. -relx => 0.5, -rely => 0.5 window resized to be larger

This next example creates two Buttons, both placed in the window with
relative coordinates:

$mw->Button(-text => "Exit", 
            -command => sub { exit })->place(-relx => 0.2, 
                                             -rely => 0.2);
$mw->Button(-text => "Exit", 
            -command => sub { exit })->place(-relx => 0.5, 
                                             -rely => 0.5);
No matter what size the window is or where other widgets are in the
screen, the two Buttons will stay in those relative locations (see
Figure 2-43).

[image: Two Buttons placed relative to the parent window]

Figure 2-43. Two Buttons placed relative to the parent window

The left window in Figure 2-43 is the default size
of the window when it was created. The right window is what it looks
like after the window was resized to make it much smaller. Notice
that the second Button placed in the window remains on top. It does
so because we are still maintaining the ordered list of widgets in
the window; the second Exit Button, placed at (0.5, 0.5), is drawn
last, so it's drawn on top of the other Button.

You can also combine the absolute and relative coordinate systems
simply by using both in the argument list. The relative coordinate
system is considered first, then the x or y value is added to that
position. The options -relx => 0.5, -x
=> -10 place the widget 10 pixels to the left of the
middle of the window.
                
            

Anchoring the Widget



Think of the child widget as a piece of paper that you want to put on
your bulletin board (the board is the parent widget). You have a tack
that you are going to use to keep the paper up on the board. You can
put the tack right through the center of the paper, in the upper-left
corner ("nw"), or in the lower-right corner
("se"). The point where the tack is going to stick
the paper to the board is the -anchor point. The
-anchor point on the widget is
"tacked" to the coordinates given by
-x, -y, and/or
-relx, -rely. The default
-anchor is "nw". Figure 2-40 shows these -anchor points
within the child widget.

It is important to know where the -anchor is,
because it will affect how we see the widget within the parent.

In Figure 2-44, almost identical
place commands were used to put the Exit Button in
the window, but the -anchor value was changed. The
left window's Button was created with this command:

$mw->Button(-text => "Exit", 
            -command => sub { exit })->place(-relx => 0.5, 
                                             -rely => 0.5);
The window on the right in Figure 2-44 used this
command:

$mw->Button(-text => "Exit", 
            -command => sub { exit })->place(-relx => 0.5,
                                             -anchor => "center",
                                             -rely => 0.5);
As with pack and grid, the
possible values for -anchor are:
'n', 'e',
's', 'w',
'center', 'nw',
'sw', 'ne', and
'se'. However, the value now refers to the child
widget instead of the position within the allocation rectangle.

[image: Different -anchor values affect where the widget is placed in the window]

Figure 2-44. Different -anchor values affect where the widget is placed in the window


Width and Height



When you use place, you can specify the width and
height of the widget in one of three ways:

	Allow the widget to determine its own size.

	Specify width and/or height in absolute measurements.

	Specify width and/or height in relative measurements (relative to the
parent widget).




To let the widgets determine their own sizes, no options are
specified. You can set the widgets' sizes with the following
options: -width and -height,
or-relwidth and
-relheight, respectively.

The -width and -height options
allow you to specify the exact width or height of the widget in a
screen distance:

-width => amount, -height => amount
Each amount is a valid screen distance (discussed earlier in this
chapter under pack). The widget will obey these
options even if it has to cut off edges of the items displayed in it.
Our Button looks quite silly on the screen when we use a
-width of 40 pixels (see Figure 2-45). 

$mw->Button(-text => "This Button Will Cause the Program to Exit", 
            -command => sub { exit })->place(-x => 0, -y => 0,
                                             -width => 40);
[image: Using -width with place]

Figure 2-45. Using -width with place

The other two options, -relwidth and
-relheight, determine the widget in relation to
the parent widget.

-relwidth => ratio, -relheight => ratio
The ratio is a floating-point number
(similar to that specified by -relx or
-rely). A value of 1.0 will make the widget as
wide (or as tall) as the parent widget. A value of 0.5 will make the
widget half as wide as the parent (see Figure 2-46).

[image: Example of the same window resized with -relwidth => 0.5, -relheight => 0.5]

Figure 2-46. Example of the same window resized with -relwidth => 0.5, -relheight => 0.5

The options -width and
-relwidth are additive when used together, and so
are -height and -relheight.


Border Options



Normally the border of the widget is used as the edge of the possible
space in the window, which means any widgets placed with either the
absolute or relative coordinate system will be placed inside the
border. This can be changed by using the
-bordermode option:

-bordermode => 'inside' | 'outside' | 'ignore'
Using 'outside' will allow the coordinate system
to use the space occupied by the border as well. A value of
'ignore' will have the coordinate system use the
space designated as the official X area. Overall, this option is
pretty useless, as you can see from the difference each makes in
Figure 2-47.

[image: -bordermode examples]

Figure 2-47. -bordermode examples

If you look very closely (get out your magnifying glass), you can see
that the 'outside' version is two pixels higher
and two pixels farther to the left than the
'inside' version. This is because with one window
manager ( fvwm), the border is defined as 2
pixels.


Methods Associated with place



The methods for place are simple and don't
allow much manipulation of the widgets.

Removing the widget



As with pack and grid, there is
a place version of the Forget
method:

$widget->placeForget();
When you use this method, the widget is removed from view on the
screen. It is also removed from the list maintained by the parent
widget.


Place information




                  
                  TheplaceInfo method returns a list of
information related to the widget:

# For easier printing:
@info = $widget->placeInfo();
print "@info";
# Or for easier fetching of info
%info = $widget->placeInfo();

## Produced these results (there are blanks where there are no values)
-x 0 -relx 0 -y 0 -rely 0 -width   -relwidth  -height  -relheight  -anchor nw

Place slaves




                  placeSlaves returns a list of the slave widgets
that are within $parent:

@widgets = $parent->placeSlaves(  );
The list looks the same as it does when it is returned from
packSlaves( )or
gridSlaves( )
                  .
                  
               



The form Geometry Manager




            The final geometry manager we
want to cover is form. Recently added into the
Perl/Tk distribution, form is a very different
geometry manager than those we've seen so far. To try and
compare it with what we already know, it behaves like a combination
of pack and place. Using
form, it is legal to overlap widgets (as you would
with place), but you can also display the widgets
relative to each other and stretch them out to fill the entire
available area and resize with the window (as you would with
pack). The combination of all these abilities
results in a powerful geometry manager.


            When using form, each
edge of a widget can be attached to something: the container's
grid, another widget, or nothing at all. You can also use springs to
push your widgets around in the window based on the strength (or
weight) of the spring. As with the other geometry managers, you can
add padding to your widget.

Let's look at the options briefly, then go into more detail on
how to use them.

Options for form




               The following
are all the legal options for form. The following
sections show you how to use these options to the best effect.

	
                     -bottom => 
                     attachment
                  
	
                        Uses the
given attachment on the bottom side of the widget.


	
                     -bottomspring => 
                     weight
                  
	
                        Uses
the given weight for a spring on the bottom side of the widget.


	
                     -fill => 'x' | 'y' | 'both' | 'none'
                  
	
                        Specifies the direction in which to
fill when springs are used. There is no default value.


	
                     -in => $master
                  
	
                        Uses $master as the
container to put the widget in. If used, the -in
option must be the first one specified.


	
                     -left => 
                     attachment
                  
	
                        Uses the given
attachment on the left side of the widget.


	
                     -leftspring => 
                     weight
                  
	
                        Uses the
given weight for a spring on the left side of the widget.


	
                     -padbottom => 
                     value, -padleft => 
                     value, -padright => 
                     value, and 
                     -padtop => 
                     value
                  
	
                        
                        Place padding on the given side of the
widget.


	
                     -padx => 
                     value
                  
	
                        Places padding on the left and right
sides of the widget.


	
                     -pady => 
                     amount
                  
	
                        Places padding on the top and bottom of
the widget.


	
                     -right => 
                     attachment
                  
	
                        Uses the given
attachment on the right side of the widget.


	
                     -rightspring => 
                     weight
                  
	
                        Uses the
given weight for a spring on the right side of the widget.


	
                     -top => 
                     attachment
                  
	
                        Uses the given
attachment on the top side of the widget.


	
                     -topspring => 
                     weight
                  
	
                        Uses the
given weight for a spring on the top side of the widget.





Attachments




               
               
               The
edge (top, bottom, left, and right) of each widget can be attached to
something else in the container. The left side can be attached to the
grid (which we'll say more about in a minute), the right to
another widget, and the top and bottom might be attached to nothing
at all. In addition to an anchor point, you can specify a positive or
negative offset from that point. An offset is a number given in
screen units. A positive offset moves the widget to the right or down
from the anchor point. A negative offset moves the widget to the left
or up from the anchor point. An attachment is comprised of either an
anchor point or an anonymous array of [
               anchor_point, offset
               ]. The examples we go through to demonstrate each
type will make this clear.

Attaching to the grid




                  There is an unseen grid in each container
widget that uses form. Don't confuse this
grid with the grid geometry manager. The default
size of form's grid is 100 x 100, and
does not change even if the container size changes. Within the
container, the left and top are both at 0, the right and bottom are
at 100. Using these numbers, we can specify a spot to attach a widget
edge to. This is different from place because
instead of using pixel points that change as the window changes size,
the size of the grid remains static. To specify a grid attachment,
use the form '%xx' replacing xx
with the grid location. If you leave out the %,
you are specifying an offset only, rather than a grid location. This
example demonstrates using a grid attachment:

foreach (1..5) {
    # Create Buttons and attach them to grid at intervals of 5
    $mw->Button(-text => $_ * 5)->form(-left => '%' . $_ * 5, 
                                       -top => '%' . $_ * 5);
}
This code creates five widgets, putting them on the grid at (5, 5),
(10, 10), (15, 15), and so on. Take a look at Figure 2-48 and you'll see what happens when we
resize this window. Notice that the widgets overlap each other and
move as the window is resized. This demonstrates that the grid size
stays the same as the window resizes, and the widgets move with the
grid. If you make the window large enough, the widgets no longer
overlap.

[image: How the form grid works (both windows have grid sizes of 100x100)]

Figure 2-48. How the form grid works (both windows have grid sizes of 100x100)


                  
                  For the previous example, we used the
-top and -left options to
specify where to locate the widgets in the window. If you call
form with no options at all, each widget is put at
(0, 0), piled one on top of the other. The order in which the widgets
are created is important in this case, because the last one created
is placed on top of the pile.

You can change the size of the widget by attaching opposite sides to
different points in the grid. The widget will resize with the window
(see Figure 2-49):

# First Button is 'normal'
$b1 = $mw->Button(-text => "small")->form(-left => '%10');
# Second Button is attached to $b1 on left, and grid 70 on right
$mw->Button(-text => "BIG")->form(-left => $b1, -right => '%70',
                                  -top => '%0', -bottom => '%80');
[image: A widget attached to the grid on two sides stretches to stay attached]

Figure 2-49. A widget attached to the grid on two sides stretches to stay attached

There are a few things to note about the code associated with Figure 2-49. We didn't specify a
-top attachment point for the small Button. The
default is always going to be 0 if there isn't a top/bottom or
left/right attachment point explicitly used with
form. Also, the -left
attachment point for the BIG Button was another widget. That's
a widget-to-widget attachment, which we'll cover next.

To shift your widget 10 pixels to the right of the grid coordinate
(5, 5), you would use an anonymous array to specify the whole
attachment:

$mw->Button(-text => "shifted right")->form(-left => ['%5', 10], 
                                            -top => '%5');
If the offset is 0, you don't need to specify it at all.

Widget-to-widget attachments




                  There are two ways to attach one widget to
another: by using the same side (both top, both bottom, and so on),
or by using opposite sides. Any widgets attached to one another must
be managed by form in the same parent container.

Here's an example of opposite side attachment:
$thisbutton->form(-left => $b1);

                  
                  You would read that in English as
"Attach the left side of $thisbutton to the
opposite (right) side of $b1." Remember the
option name you are using (-left,
-right, -top, or
-bottom) applies to the widget on which you are
currently calling form. If you want a bit of space
between two widgets, add an offset to the -left
attachment point:

$b1 = $mw->Button(-text => "small")->form(-left => '%10');
# Second Button is attached to $b1 on left, and grid 70 on right
$mw->Button(-text => "BIG")->form(-left => [$b1, 5], -right => '%70',
                                  -top => '%0', -bottom => '%80');
By changing the argument to -left (in bold), we
add a bit of space between the two widgets, as shown in Figure 2-50. 

[image: Using an offset with a widget attachment]

Figure 2-50. Using an offset with a widget attachment


                  
                  What if
we want to line up the left sides of a bunch of widgets? We use a
same-side attachment. (In the docs this is called Parallel Side
Attachment). To tell form you are using a
same-side attachment, add the '&' character as
the first item in an anonymous array.

$b1 = $mw->Button(-text => 'top widget')->form(-left => '%10');
$b2 = $mw->Button(-text => 'bottom widget')->form(-left => ['&', $b1], 
	-top => $b1);
$mw->Button(-text => 'bottom widget(2)')->form(-left => ['&', $b1, ], 
	-top => [$b2, 10]);
This code creates three Buttons, all left aligned, with the third
Button 10 pixels from the bottom of the one above it. We are lining
up both the second and third Button to $b1. See
what this looks like in Figure 2-51.

[image: Using same-side and opposite-side attachments together]

Figure 2-51. Using same-side and opposite-side attachments together

What if we want to stretch all three widgets to the same length? With
pack we'd just use -fill
                  => 'y' inside the container.
With form, we add -right
                  => '%100' to the options of
each form call. This will stretch each widget all
the way to the right edge of the container, as shown in Figure 2-52. 

[image: Stretching widgets to the same point in the container]

Figure 2-52. Stretching widgets to the same point in the container

Using form this way has one disadvantage: if you
wanted to put something to the right of all those Buttons,
you'd be better off sticking all three widgets in a Frame and
treating them as one. A way around this is to use an attachment to
the grid of -right =>
                  '%80' on all three widgets, but crazy things can
happen if the window is sized smaller than you expect.

Here's an example of centering a widget directly in the
container:

# Center a widget across the whole screen:
$w = -($a->reqwidth( )/2);
$h = -($a->reqheight( )/2);
print "W: $w, H: $h\n";
$a->form(-top => ['%50', $h], -left => ['%50', $w]);

                  
                  We use reqwidth and
reqheight to find out how large the widget should
be, then use those values as offsets to the center grid position
'%50' on both the top and left. In order to shift
the widget correctly, we make the offsets negative. It isn't
necessary to specify -right and
-bottom edges.


Attaching to nothing



You can state that a widget is attached to nothing at all on the
specified side by using 'none' as the attachment
value. Since this is the default for any side attachments that
aren't listed, it won't be necessary to use this very
often.
                   
               


Springs




               As this
book was being written, springs were not fully implemented.[2] They are
supposed to act as forces on each side of the widget to adjust the
position of the widget within its container. If the widget is
attached to another widget, the result can be a widget that resizes
properly, but doesn't overlap as things move around.


               If you
use a spring, it has a weight. The weight
doesn't really matter except in relation to any other spring
weights you are using in your parent container. If each spring has a
weight of 1, they are all pushing with the same amount of force. The
same can be said if every spring has a weight of 2. If one spring has
a weight of 1, and another a weight of 2, then the second spring is
twice as powerful as the first spring.

To discover what springs really do to your widgets, here's a
"Play with form" widget displaying program:

use Tk;

$mw = MainWindow->new(-title => 'Play w/form');


# Create a Frame at the bottom of the window to use 'form' in
$f = $mw->Frame(-borderwidth => 2, -relief => 'groove')
	->pack(-side => 'bottom', -expand => 1, -fill =>'both');

# Display the Button in the default position to start
$button = $f->Button(-text => "Go!", -command => \&reForm)->form;

# Use grid to create the Entry widgets to take our options:
$f1 = $mw->Frame->pack(-side => 'top', -fill => 'x');
$f1->Label(-text => '-top')->grid($f1->Entry(-textvariable => \$top),
	$f1->Label(-text => '-topspring'),  
	$f1->Entry(-textvariable => \$topspring), 
	-sticky => 'w', -padx => 2, -pady => 5);

$f1->Label(-text => '-bottom')->grid($f1->Entry(-textvariable => \$bottom), 
	$f1->Label(-text => '-bottomspring'), 
	$f1->Entry(-textvariable => \$bottomspring), 
	-sticky => 'w', -padx => 2, -pady => 5);

$f1->Label(-text => '-left')->grid($f1->Entry(-textvariable => \$left),
	$f1->Label(-text => '-leftspring'), 
	$f1->Entry(-textvariable => \$leftspring), 
	-sticky => 'w', -padx => 2, -pady => 5);

$f1->Label(-text => '-right')->grid($f1->Entry(-textvariable => \$right),
	$f1->Label(-text => '-rightspring'), 
	$f1->Entry(-textvariable => \$rightspring), 
	-sticky => 'w', -padx => 2, -pady => 5);

# Add this Button in case the options we put in causes the 'formed' Button 
# to go off screen somewhere.
$f1->Button(-text => "Go!", -command => \&reForm)
	->grid('-', '-', '-', -pady => 5);

MainLoop;

sub reForm
{
	print "top => $top\t";
	print "topspring => $topspring\n";
	print "bottom => $bottom\t";
	print "bottomspring => $bottomspring\n";
	print "left => $left\t";
	print "leftspring => $leftspring\n";
	print "right => $right\t";
	print "rightspring => $rightspring\n";
	print "-----------------------------\n";
	
	# Remove Button from container for now
	$button->formForget;
	
	my @args = (  );

	if ($top ne '') { push (@args, ('-top', $top)); }
	if ($bottom ne '') { push (@args, ('-bottom', $bottom)); }
	if ($right ne '') { push (@args, ('-right', $right)); }
	if ($left ne '') { push (@args, ('-left', $left)); }
	if ($topspring ne '') { push (@args, ('-topspring', $topspring)); }
	if ($bottomspring ne ''){push (@args, ('-bottomspring', $bottomspring));}
	if ($rightspring ne '') { push (@args, ('-rightspring', $rightspring)); }
	if ($leftspring ne '') { push (@args, ('-leftspring', $leftspring)); }
	print "ARGS: @args\n";

	# Put Button back in container using new args
	$button->form(@args);

}
You can't enter anonymous arrays as values in this application,
but you can play with the basic way of attaching to a grid and with
the springs. Take a look at Figure 2-53 to see what
the widget looks like after the options are changed.

[image: Using the Play with form program to test options]

Figure 2-53. Using the Play with form program to test options

Using the top and bottom springs pushes the widget around a bit: if
you enter 1 for both -topspring and
-bottomspring, the height of the widget goes back
to its default size and becomes centered in the screen.


form Methods




               The following sections summarize
form's methods.

Changing the grid size




                  
                  
                  You can change the grid at any time by
calling formGrid with new x and y sizes. You can
also determine the current size of the grid.

$parent->formGrid(200, 200);
($x, $y) = $parent->formGrid;

Removing a widget from a container




                  
                  
                  To
take a widget out of its container, call
$widget->formForget. Any attachments defined
for this widget no longer apply.


Options info for form




                  
                  Just like all the other geometry managers,
form has an Info method to get
detailed information about how a widget was put in its parent
container:

%forminfo = $widget->formInfo;
$top_value = $widget->formInfo(-top);

What's managed by form?




                  
                  
                  To
get a list of all the widgets managed by a parent container using
form, call formSlaves on the
parent. The order in the list is the same as how the widgets were
added into the parent.

@widgets = $parent->formSlaves;
print "Widgets in $parent are @widgets\n";

Circular dependency check




                  
                  You need to be careful that you
don't end up with a circular reference when doing widget
attachments. To check your setup, you can call
$parent->formCheck. This will return true if it
finds a circular dependency and false if it doesn't.
               




[2] It appears that top and bottom springs work, but left and right
don't. Hopefully this will be fixed soon, as springs make
form more powerful.



Geometry Management Summary




            You
now know more about the different geometry managers than you'll
ever need to write a successful Perl/Tk application. Here are some
helpful hints on deciding which geometry manager to use:

	
                  pack
                   is
good for general purpose use and will be your choice about 95% of the
time.


	
                  grid
                   is
perfect for those situations in which you would like to create a
columnar layout similar to a spreadsheet. Options allow you to change
the sizes of rows and/or columns easily.


	
                  place
                   is
most useful when you want your widget to stay in a position or size
that is relative to the widget that created it. When used correctly,
it can be very powerful.


	
                  form is powerful, but difficult to get used to;
not for the faint of heart. Check future releases of the Tk module
for updates to this geometry manager.




No matter which manager you use, take the time to get the widgets on
your window where they belong (or more likely, where you want them).
There's nothing more unsettling than a Button that looks like
it just doesn't belong in the window.

As you read through this book, you'll notice that some of the
option names for the geometry managers are also option names when you
are creating or configuring a widget type. For example, you can
specify the -width of a Button without using
place. Always keep in mind the context in which
the option is used. Sometimes the functional difference is very
subtle.


Chapter 3. Fonts



Every computer system uses fonts. In Perl/Tk applications, you can
change any of the fonts for items inside the
application, but not the fonts used in the window decoration for
titles (which are handled by the window manager). In this chapter,
we'll show you how to use fonts in your Perl/Tk application.


         
         
         What
fonts do you have available? For MS Windows users, the available
fonts can be found in the Font control panel. Users of the X Window
System can get a font list by running xlsfonts.
The font used in this chapter is Linotype Birka. While you
can't do anything about the font used in this book, you can do
something about the fonts in the applications you create or run in
Perl/Tk.

The simplest way of altering an application's font is changing
the base font for the entire application. You can do that with any
Perl/Tk application by using a command-line option:

perl myTkApp.pl -font "Times 12"

         Using the
-font command-line option doesn't require
any changes to your Perl script. The -font option
works because of the way Tk::CmdLine works, described in Chapter 16. Note that you specify the
-font option after the name of the program to run.
As long as you haven't explicitly specified the font for any
widgets in your application, all widgets will use the new font.

To change the font for only some widgets, you can use the option
database, described in Chapter 16. For example, if
you wanted to change only the font for Text widgets in your
application, specify *text*font=Courier
         16 in the option database.

Experimenting with Fonts



You don't generally want to hardcode font specifications in your
programs. Simply put, it prevents your users from customizing your
applications. There are extenuating circumstances, though; you might
have an HP calculator that has a specific look that shouldn't
be changed (see Chapter 15). Creating such a
specific look might require one or more particular fonts.


            
            One way to determine what font to use is
to write a program using the fontFamilies method
that displays various font specifications. So, before we get into the
details of creating a font definition, let's look at a program
that lets us play around with the fonts on our system. This program
is useful no matter what operating system you're on.

use Tk;
use Tk::BrowseEntry;
use strict;

my $mw = MainWindow->new(-title => 'Font Viewer');
my $f = $mw->Frame->pack(-side => 'top');

my $family = 'Courier';
my $be = $f->BrowseEntry(-label => 'Family:', -variable => \$family,
  -browsecmd => \&apply_font)->pack(-fill => 'x', -side => 'left');
$be->insert('end', sort $mw->fontFamilies);

my $size = 24;
my $bentry = $f->BrowseEntry(-label => 'Size:', -variable => \$size, 
  -browsecmd => \&apply_font)->pack(-side => 'left');
$bentry->insert('end', (3 .. 32));

my $weight = 'normal';
$f->Checkbutton(-onvalue => 'bold', -offvalue => 'normal', 
  -text => 'Weight', -variable => \$weight, 
  -command => \&apply_font)->pack(-side => 'left');

my $slant = 'roman';
$f->Checkbutton(-onvalue => 'italic', -offvalue => 'roman', 
  -text => 'Slant', -variable => \$slant, 
  -command => \&apply_font)->pack(-side => 'left');

my $underline = 0;
$f->Checkbutton(-text => 'Underline', -variable => \$underline, 
  -command => \&apply_font)->pack(-side => 'left');

my $overstrike = 0; 
$f->Checkbutton(-text => 'Overstrike', -variable => \$overstrike, 
  -command => \&apply_font)->pack(-side => 'left');

my $stext = 'Sample Text';
my $sample = $mw->Entry(-textvariable => \$stext)->pack(-fill => 'x');

&apply_font;

MainLoop;

sub apply_font {
  # Specify all options for font in an anonymous array
  $sample->configure(-font => 
    [-family => $family,
     -size => $size,
     -weight => $weight,
     -slant => $slant,
     -underline => $underline,
     -overstrike => $overstrike]);
}

            Figure 3-1 shows what the window looks like if we
select Garamond, size 24, with slant and overstrike.

[image: Font viewer]

Figure 3-1.  Font viewer


            Those of you used to a Unix system
will recognize this type of font viewer, since there is something
similar that comes with X, called xfontsel.

In our font viewer, we can see the changes to the font are applied
using the apply_font subroutine. We specify all
aspects of the font so the user has the choice of changing any part
of the font. Let's talk about those different parts.



Dissecting a Font




            In
the font viewer, we see that the Entry's font is changed with
the -font option using an anonymous array. From
this we know a font consists of the following things:

	Family
	
                     The actual
name of the font, e.g., 'Courier',
'Times', and so on.


	Size
	
                     
                     The size of the font in points. The
larger the size, the larger the text displayed on the screen. A point
is 1/72 of an inch. Negative values are interpreted as pixels.


	Weight
	
                     Determines if
the font is shown boldor not. The
value 'normal' means it is not shown bold, and
'bold' makes the font thicker.


	Slant
	
                     Shows straight
up and down if 'roman' is used, and slanted if
'italic' is used.


	Underline
	
                     If the value used with
-underline is true, the text will be underlined.
If false, the text will not be underlined.


	Overstrike
	
                     If true,
a line will be drawn through the center of the text.





            
            
            If you are used to working with fonts on a
Unix system, you are probably familiar with X Logical Font
Descriptions (XFLD). This is the dash-delimited format used for fonts
under X, for example:

*-helvetica-bold-r-*-*-*-240-*-*-*-*-*-*
This font description indicates a 24-point bold Helvetica font with
Roman slant. The field order is as follows:
-foundry-family-weight-slant-sWdth-adstyl-pixelsize-pointsize-resx-resy-spacing-avgWidth-registry-encoding.

When specifying a font in XLFD notation, an asterisk means you
don't care what is used for that value, and the system will
choose a default for you.

While a full description of X fonts is beyond the scope of this book,
there are a few things you should know. First, it is entirely
acceptable to specify a font in XLFD notation under either Unix or
Windows. The code is:

font => '-*-helvetica-bold-r-*-*-*-240-*-*-*-*-*-*'

            When you
use this format under Unix, you get exactly what you ask for. Every
field is honored. Under Windows, only family, weight, slant,
pixelsize, and pointsize are honored. All the other fields are
ignored.


            
            
            If you
would like to learn more about fonts under X, a good beginner's
guide and bibliography may be found in "The X Window User
HOWTO" by Ray Brigleb. This document may be found in many, many
places on the Web, but the definitive location is http://www.linuxdoc.org/HOWTO/XWindow-User-HOWTO.html.


Using Fonts




            Now
that we know what comprises a font, let's look at a few ways we
can specify them in code.


            We
simplify things in our Perl/Tk applications by being able to create a
single name that refers to a combination of family, size, weight,
slant, underline, and overstrike:

$code_font = $mw->fontCreate('code', -family => 'courier',
                             -size => 12);
Once we have created our new font, you can refer to the font by the
variable $code_font or by the name,
'code':

$mw->Button(-text => "Show Code", -font => 'code');
$mw->Button(-text => "Show Code2", -font => $code_font);
It is much simpler to specify all the desired font options once and
refer to them using the name or variable later in the program. If you
don't want to use a name for the font, don't specify it;
the system will generate a name for you automatically.

$code_font = $mw->fontCreate(-family => 'courier',
                             -size => 12);

            Once
the font is created, you can change any of its settings using the
fontConfigure method, using the font name or
reference as the first argument:

$mw->fontConfigure($code_font, -family => 'Verdana');
The changes will take effect immediately on any widgets using that
font, making it very useful for on-the-fly changes.


            
            The -font option will
also accept an anonymous array containing the right parts, with or
without the identifiers:

-font => ['courier', '14', 'bold']
# The same thing, but more verbose:
-font => [-family => 'courier',
          -size => '14',
          -weight => 'bold']
The second way is much more verbose, and easier to read, but those of
us who prefer to keep our code small and compact might want to stick
with the first method. You must specify at minimum the family name;
all other specifications are optional.

If creating an anonymous array isn't to your liking, try just
using a string containing the relevant parts:

-font => "courier 14 bold"
-font => "courier 12 bold italic"
-font => "{courier} 14"
-font => "{Calisto MT} 18 bold italic overstrike"
There are a few restrictions when using this specification. The
family name must always come first, followed by the (optional) size,
and any of weight, slant, and so on. If the family name has a space
in it, you must put it between curly braces so the font parser can
find the full family name. You can put any family name in curly
braces, not just those that have spaces in them; if you like this way
of specifying fonts, it might be best to always include the curly
braces.

System Fonts




               
               
               
               
               In addition to the fonts that are listed
with xlsfonts or shown in the Font Control Panel,
you can also specify fonts referred to as system
fonts. Since these fonts are operating system specific, you will not
get the same result from machine to machine, unless they happen to be
running the same operating system. Table 3-1 lists
system fonts on each of the popular operating system platforms.

Table 3-1. System fonts
	
                           Platform

                        	
                           System fonts

                        
	
                           Unix

                        	
                            (Use xlsfonts for a complete listing)

                        
	
                           Windows

                        	
                            system, systemfixed, ansi, ansifixed, device, oemfixed 

                        
	
                           Macintosh

                        	
                            system, application

                        





Using Fonts Dynamically




            Let's
look at a program that creates fonts dynamically. This code will
display each selected font in a window much the way Microsoft Windows
does when you look at the Font control panel. To display the font in
its different sizes, we simply use the ROText widget so the text is
read-only (see Chapter 8 for further information
on ROText widgets). The font changes are applied in the
show_font sub using tags.

use Tk;
require Tk::TList;
require Tk::ROText;
use strict;

my $mw = MainWindow->new(-title => "Fonts");
$mw->minsize(700,400);
my $tl = $mw->Scrolled("TList", -font => ['Arial', '12'], -command => \&show_font)->
pack(-fill => 'both', -expand => 1);

# using a tlist, we have to insert each item individually
foreach (sort $mw->fontFamilies)
{
		$tl->insert('end', -itemtype => 'text', -text => $_);
}

MainLoop;

# called when user double clicks on a font name in the tlist.
sub show_font
{
		my ($index) = @_;
		my $name = $tl->entrycget($index, -text);
		my $top = $mw->Toplevel(-title => $name);
		my $text = $top->Scrolled("ROText", -wrap => 'none')
      ->pack(-expand => 1, -fill => 'both');
		
		$text->tagConfigure('number', -font => ['courier', '12']);

    # since we don't know what font they picked, we dynamically
    # create a tag w/that font formatting
		$text->tagConfigure('abc', -font => [$name, '18']);
		$text->insert('end', "abcdefghijklmnopqrstuvwxyz\
nABCDEFGHIJKLMNOPQRSTUVWXYZ\n1234567890.;,;(*!?')\n\n", 'abc');
		
		foreach (qw/12 18 24 36 48 60 72/)
		{
			$text->tagConfigure("$name$_", -font => [$name, $_]);
			$text->insert('end', "$_ ", 'number');
 			$text->insert('end', 
 "The quick brown fox jumps over the lazy dog. 1234567890\n", "$name$_");
		}
}
 
            Figure 3-2 and Figure 3-3
show the resulting windows.

[image: MainWindow in our control panel-like Font viewer]

Figure 3-2.  MainWindow in our control panel-like Font viewer

[image: An individual font (Courier) viewed in a Toplevel widget]

Figure 3-3.  An individual font (Courier) viewed in a Toplevel widget


Font Manipulation Methods



Once you've created a font using fontCreate,
you can use the following methods.


            
            For a description of the font's
attributes (some or all), use fontActual to query
the font:

$mw->fontCreate('bigfont', -family => 'Arial', -size => 48);

%big = $mw->fontActual('bigfont');
print %big;
# prints:
-size 48 -overstrike 0 -underline 0 
-weight normal -slant roman -family Arial

$size = $mw->fontActual('bigfont', -size);
print $size;
#prints: 
48

            
            To change (or query) a property of a font
once it has been created, use fontConfigure:

if ($mw->fontConfigure('bigfont', -size) < 24) {
  $mw->fontConfigure('bigfont', -size => 48);
}

# same as $mw->fontActual('bigfont');
%bigfont = $mw->fontConfigure('bigfont');

            
            If you'd like to delete a font
definition, use fontDelete:

$mw->fontDelete('bigfont');
If you delete a font that is being used, the widgets using it
won't change what they display. They display whatever font they
were last. If you try to manipulate the font programmatically after
it's been deleted, you will get an error.


            To get a
list of all the font families available on your system, use
fontFamilies:

@families = $mw->fontFamilies;

            
            To get a list of the currently defined
named fonts on your system, use fontNames:

@definedfonts = $mw->fontNames;
The fontNames method returns a list of object
references to Font objects. The list will be empty if there
aren't any fonts defined on your system. Keep in mind this list
contains only those fonts defined using the
fontCreate method.


            
            
            If you want to determine how much
horizontal space a piece of text will take up
with a given font, use fontMeasure. The answer is
given in pixels. Don't count on this figure to be the exact
size; it's more of an estimate.

print $mw->fontMeasure('bigfont', "SHORT"), "\n";
225
print $mw->fontMeasure('bigfont', "MUCH LONGER"), "\n";
480

            
            
            
            
            
            For those of you who really understand
the different ways you can measure a font, you'll be interested
in the fontMetrics method. The vertical options it
describes are -ascent (very top to baseline),
-descent (very bottom to baseline),
-linespace (separation between lines of text), and
-fixed (whether or not the font is proportional).

$ascent = $mw->fontMetrics('bigfont', -ascent);
%metrics = $mw->fontMetrics('bigfont');
print %metrics;

-linespace 72 -descent 14 -ascent 58 -fixed 0
One Last Example



Just for fun we'll look at one last example. This program will
display a banner in the selected font that continually rotates the
text within it (see Figures Figure 3-4 and Figure 3-5). 

[image: Banner program showing the font configuration widgets]

Figure 3-4.  Banner program showing the font configuration widgets

[image: Banner program without font configuration widgets]

Figure 3-5.  Banner program without font configuration widgets

Here's the code for the banner program:
use Tk;
use strict;

# initial banner text. Entry is not read-only
my $str = "AaBbCcDdEeFfGgHhIiJjKkLlMmNnOoPpQqRrSsTtUuVvWwXxYyZz0123456789";

my $mw = MainWindow->new;
my $lframe = $mw->Frame->pack(-fill => 'both', 
  -side => 'left', -expand => 1);
my $lb = $lframe->Scrolled("Listbox", -scrollbars => "e", 
  -height => 3)->pack(-fill => 'both', -expand => 1, -side => 'top');

$lb->insert('end', sort $mw->fontFamilies);

# Button that will pop the config widgets in and out
my $hidebutton = $mw->Button(-text => ">")->pack(-side => 'left', 
  -fill => 'y');
$hidebutton->configure(-command => 
  sub { 
		if ($hidebutton->cget(-text) eq ">") {
		  $lframe->packForget; $hidebutton->configure(-text => "<")
		} else {
		  $lframe->pack(-before => $hidebutton, -fill => 'both',
        -side => 'left', -expand => 1); 
		  $hidebutton->configure(-text => ">");
		}
  }, -font => "courier 8");

my $entry = $mw->Entry(
		-textvariable => \$str,
		-width => 12, 
		-font => "{Comic Sans MS} 72",
		-relief => 'raised',
		-highlightthickness => 0,
		)->pack(-expand => 1, -fill => 'x', -side => 'left');
		
$lb->bind("<Button>", sub { $entry->configure(
   -font => "{". $lb->get($lb->curselection) . "} 72"); });

my $repeat_id = $mw->repeat(300, \&shift_banner);

my $f = $lframe->Frame->pack(-side => 'bottom', -fill => 'y');
my $start_button;
$start_button = $f->Button(-text => "Start", 
  -command => sub { 
    $repeat_id = $mw->repeat(300,\&shift_banner);
    $start_button->configure(-state => 'disabled'); }, 
  -state => 'disabled')->pack(-side => 'left', -padx => 3);
my $stop_button = $f->Button(-text => "Stop", -command => sub {
    $repeat_id->cancel(  ); 
    $start_button->configure(-state => 'normal'); }
  )->pack(-side => 'left', -padx => 3);

MainLoop;

# Causes text to be wrapped around in entry
sub shift_banner {
		my $newstr = substr($str, 1) . substr($str, 0, 1);
		$str = $newstr;
}


Chapter 4. 
Button, Checkbutton, and Radiobutton Widgets




         Almost all Perl/Tk applications use
Buttons in one way or another. There are three different types of
Button widgets available in the standard Perl/Tk set: Button,
Checkbutton, and Radiobutton. This chapter covers all three types of
Buttons and includes examples of where you might use one over the
other.


         
         Each of the Buttons we cover in this
chapter look different, primarily in their use of
selection indicators. The Button widget
doesn't use indicators at all, but Checkbutton and Radiobutton
widgets use them to indicate whether the Button has been selected or
not. The Button widgets are:

	Button
	A plain Button, shown in Figure 4-1. The user can
press it and usually an immediate action results.




[image: Button widget]

Figure 4-1. Button widget

	Checkbutton
	
                  A
Checkbutton, shown checked in Figure 4-2. When
checked or unchecked, only the visual representation is changed; the
state is not validated until later in the program. Checkbuttons can
be used singly or in groups.




[image: Checkbutton widget]

Figure 4-2. Checkbutton widget

	Radiobutton
	
                  A
Radiobutton, shown unchecked in Figure 4-3. When
checked or unchecked, only the visual representation is changed; the
state is not validated until later in the program. Radiobuttons are
always used in groups of two or more.




[image: Radiobutton widget]

Figure 4-3. Radiobutton widget


         A Button is one of the simplest Perl/Tk
widgets: the user presses it and something immediately happens. The
label of the Button should make the action clear; for example, text
such as Quit, Save, or Print gives the user a good idea of what will
happen when she clicks the Button. After the Button has been clicked,
it will look exactly the same as before, unless programmed to change
text or color.


         Checkbuttons are for when you want to
select none, some, or all items. For example, a shopping list might
consist of Bread, Milk, Soda, and Eggs. Select none if you
don't need any of them, or select all if you're out of
everything. As the user clicks on each Checkbutton, the selection
indicator will be filled in or left blank as appropriate (see Figure 4-4). 

[image: Checkbuttons used for a shopping list]

Figure 4-4. Checkbuttons used for a shopping list

After a user presses the Purchase Button, the code examines the value
of each Checkbutton's variable to decide what tasks to perform
next.


         Radiobuttons, on the other hand, all assign
the same variable and are necessarily related. Radiobuttons are used
in situations when you must make a choice between items, such as on a
multiple-choice exam, as shown in Figure 4-5.

[image: Radiobuttons]

Figure 4-5. Radiobuttons

Because each Radiobutton in a group is associated with the same
variable, you are forced to select one and only one choice in that
group. If the default choice is always E and you click on D, E is
unselected automatically.

Creating Button Widgets




            As with
any widgets you create, you call a method from the parent widget that
matches the name of the widget:

$button = $mw->Button->pack;
$rb = $mw->Radiobutton->pack;
$cb = $mw->Checkbutton->pack;
These are unrealistic examples as you will most likely use some
options when creating each different Button type:

# Create a Button widget
$mw->Button(-text => 'Go', -command => \&go_go_go)->pack;

# Create a Checkbutton
$cb = $mw->Checkbutton(-text => 'Red', -onvalue => 'Red', 
	-offvalue => '')->pack;

# Create three Radiobuttons in Frame widget $f1
# Link them using $favcolor
foreach (qw/red blue green/) {
    $f1->Radiobutton(-text => $_, -variable => \$favcolor, 
        -value => $_)->pack(-anchor => 'w');
}
We'll explain the options used in the previous examples in
upcoming sections. In particular, -command expects
a callback, which we'll mention briefly in Section 4.9, but we won't fully describe
until Chapter 15.


            The only time you might not want to save a
reference is when you create a Button, set the text, and set a simple
callback for it all at once:

$mw->Button(-text => 'Quit', -command => sub { print 'Bye!'; exit; })->pack;


Standard Options for Each Button Type




            Before we get into all the options
available for each of the Button widgets, let's take a look at
the most common ones.


            
            When creating a Button, use the
-text and -command options. The
-text option lets the user know what the Button is
for, and the -command option makes something
happen when the user clicks the Button.

$b = $mw->Button(-text => 'Exit', -command => sub { exit; } )->pack;

# Use the same sub for many Buttons
$b = $mw->Button(-text => 'Red', -command => [\&change_color, 'red'])->pack;
$b = $mw->Button(-text => 'Blue', 
	-command => [\&change_color, 'blue'])->pack;
$b = $mw->Button(-text => 'Green', 
	-command => [\&change_color, 'green'])->pack;

            
            When creating Checkbuttons, you use
-variable in addition to -text.
Using -variable gives you an easy way to find out
whether the Checkbutton is checked. (You will rarely use
-command with a Checkbutton):

$mw->Checkbutton(-text => 'Print Header', -variable => \$print_header);

sub print_document {
	if ($print_header) {
		# Code to print header here...
	}
}
The value stored in $print_header is 1 or 0. A
simple test will tell you if the Checkbutton was checked.


            
            
            When
creating Radiobuttons, we always create more than one and use the
-text, -variable, and
-value options:

$group1 = 100; # set default value
foreach (qw/1 10 100 10000 100000 1000000/) {
	$mw->Radiobutton(-text => '$' . $_, -variable => \$group1, 
		-value => $_)->pack(-side => 'left');
}

print "User selected: $group1";
The variable $group1 relates all of the
Radiobuttons, making it so the user can select only one at a time.
Each Radiobutton must be given a -value to store
in $group1 (there is no default).


Table of Options for Button-Type Widgets




            
            
            The
Button widgets share almost all of the same options. Table 4-1 shows a complete list of options and which
widget they apply to. We'll cover these options in more detail
as we explore what Buttons can do.


            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            
            In the details following Table 4-1, all information applies equally to Buttons,
Checkbuttons, and Radiobuttons unless explictly stated otherwise.

Table 4-1. Options for Button-type widgets 
	
                        Option

                     	
                        Button

                     	
                        Checkbutton

                     	
                        Radiobutton

                     
	
                        	
                                 -activebackground => 
                                 color
                              
	Sets the color the background should be when the mouse cursor is over
the Button. A color is a text string such as
"red".

                                 -activeforeground => 
                                 color
                              
	Sets the color the text should be when the mouse cursor is over the
Button.

[image: image with no caption]


                        

                     
	
                        	
                                 -anchor => 'n' | 'ne' | 'e' | 'se' | 's' | 'sw' | 'w' | 'nw' |
                                  'center'
                              
	Causes the text to stick to the specified position in the Button.

                                 -background => 
                                 color
                              
	Sets the background of the Button to color.

                                 -bitmap => 'bitmapname'
                              
	Sets default bitmap or the location of a bitmap file (with
@ in front of path). See Chapter 17 for more details.

                                 -borderwidth => 
                                 amount
                              
	Changes the width of the edge drawn around the Button and the
thickness of the indicator. Emphasizes the -relief
of the Button.

                                 -command => 
                                 callback
                              
	Indicates a pointer to a function that will be called when the Button
is pressed.

                                 -cursor => 'cursorname'
                              
	Indicates that the mouse cursor will change to
'cursorname' when over the Button. See Chapter 23 for details.

                                 -disabledforeground => 
                                 color
                              
	Sets the color the text should be when the Button is disabled
(-state will be 'disabled').

                                 -font => 'fontname'
                              
	Changes the font of all text on the Button.

                                 -foreground => 
                                 color
                              
	Changes the text color to color.

                                 -height => 
                                 amount
                              
	Sets the height of the Button in characters if text is displayed and
the screen distance if an image or bitmap is displayed.

                                 -highlightbackground => 
                                 color
                              
	Sets the color of the area behind the focus rectangle (shows when
widget does not have focus).

                                 -highlightcolor => 
                                 color
                              
	Sets the color of the focus rectangle (shows when widget has focus).

                                 -highlightthickness => 
                                 amount
                              
	Sets the thickness of the highlight rectangle around the Button;
indicates focus.

                                 -image => $imgptr
                              
	
                                    $imgptr is a pointer to an Image object made with
any supported image format. See Chapter 17 for
details.





                     	
                        
                           
[image: image with no caption]


                        

                     	
                        
                           
[image: image with no caption]


                        

                     	
                        
                           
[image: image with no caption]


                        

                     
	
                        	
                                 
                                    1
                                 
                              
	Determines whether to display the indicator.

                                    'center'
                                 
                              
	Sets the direction against which multiline text will justify.

                                 -offvalue => 
                                 newvalue
                              
	Sets the value used when the Button is off. Must be a scalar. Default
is 0.

                                 -onvalue => 
                                 newvalue
                              
	Sets the value used when the Button is on. Must be a scalar. Default
is 1.

                                 -padx => 
                                 amount
                              
	Adds extra space to the left and right side of the Button inside the
Button edge.

                                 -pady => 
                                 amount
                              
	Adds extra space to the top and bottom of the Button inside the
Button edge.





                     	
                        
                           
[image: image with no caption]


                        

                     	
                        
                           
[image: image with no caption]


                        

                     	
                        
                           
[image: image with no caption]


                        

                     
	
                        	
                                 -relief =>'flat'|'groove'|'raised'|'ridge'|'sunken'|'solid'
                              
	Changes the type of edges drawn around the Button. Default for
Checkbutton and Radiobutton is 'flat'.

                                 -selectcolor => 
                                 color
                              
	Sets the color of the indicator when on.

                                 -selectimage => 
                                 imgptr
                              
	Indicates the image to display instead of text when Button is on.
Ignored if -image is not used.

                                 -state => 'normal' | 'disabled' | 'active'
                              
	Indicates the Button's state of responsiveness. If set to
'disabled', the Button does not respond.

                                 -takefocus => 0 | 1 | undef
                              
	Indicates that the Button will never get focus (0), always get focus
(1), or let the application decide (undef).

                                 -text => 'text'
                              
	Sets the text string displayed on the Button.

                                 -textvariable => \$variable
                              
	Points to a variable containing text to be displayed in Button.
Button text will change as $variable does.

                                 -underline => 
                                 n 
                              
	Underlines the nth character in the text
string. Allows keyboard input via that character when Button has the
focus. It's important to note that the character is only
underlined; any behavior must be supplied by the programmer via a
bind command. See Chapter 15 for
binding details.

                                 -value => 
                                 newvalue 
                              
	Sets the value assigned to $variable (with
-variable option) when this Radiobutton is
selected. Default is 1.

                                 -variable => \$value
                              
	Associates the on/off values with $variable.

                                 -width => 
                                 amount
                              
	Sets the width of the Button in characters if text is displayed and
as a screen distance if an image or bitmap is displayed.

                                 -wraplength => 
                                 amount
                              
	Sets the screen distance for the maximum amount of text displayed on
one line. The default is 0, which means that text is not wrapped at
word boundaries, only at line breaks (newlines).

End of sample




    To search for additional titles please go to 

    
    http://search.overdrive.com.   

